code:
;오늘팬티 처리의 세련
;ARG가 입을 수가 있는 팬티를 추첨한다
;원래의 오늘의팬티 함수라고 처리 상한이 없기 때문에 셔플 한다
;FISHER_YATES_SHAFFLE 사용하려고 하니 F함수 내부에서 CALL 할거야 무리였다이다
@오늘의팬티(ARG)
#FUNCTION
#LOCALSIZE 2
#LOCALSSIZE 1
#DIM TODAYS_PANTS
#DIM RAND_PANTS, MAXPANTS
#DIM P_TABLE

VARSET RAND_PANTS , 0
FOR LOCAL:0 , 0 , MAXPANTS
	RAND_PANTS:(LOCAL:0) = LOCAL:0
NEXT
FOR LOCAL:0 , 0 , MAXPANTS - 1
	LOCAL:1 = RAND(LOCAL:0 , MAXPANTS)
	SWAP RAND_PANTS:(LOCAL:0), RAND_PANTS:(LOCAL:1)
NEXT

FOR P_TABLE, 0, MAXPANTS
	IF GET_INT(ARG, "하반신속옷_비켜놓기가능", RAND_PANTS:P_TABLE, "입다")
		TODAYS_PANTS = RAND_PANTS:P_TABLE
		BREAK
	ELSE
		CONTINUE
	ENDIF
	SIF P_TABLE == MAXPANTS - 1
		THROW %CALLNAME:ARG%의 팬티 추첨에 실패
NEXT
RETURNF TODAYS_PANTS

;호환용
@PANTSNAME(ARG, CHARA)
#FUNCTIONS
#DIM CHARA

CALLF 하반신속옷체크(CHARA)
RETURNF GET_STR(CHARA, CLOTHES_PARTS_TO_CATEGORY("하반신속옷2"), ARG, "이름")

@PANTS_DESCRIPTION(ARG, CHARA)
#FUNCTIONS
#DIM CHARA
RETURNF GET_STR(CHARA, CLOTHES_PARTS_TO_CATEGORY("하반신속옷2"), ARG, "묘사")

@PANTS_GET(ARG, ARGS)
#DIM 종류
종류 = EQUIP:ARG:하반신속옷2
CALL PANTS_GETTING(ARG, 종류)

CFLAG:ARG:노팬티 = 1
EQUIP:ARG:하반신속옷2 = 0
CALL CLOTHES_SETTING_TRAIN(ARG)

SIF CFLAG:ARG:파자마 == 1
	CFLAG:ARG:파자마팬티 = 0

SIF ARGS == "받는다"
	CFLAG:ARG:노팬티 = 2

@PANTS_GETTING(ARG, 종류)
#DIM 종류
#DIM 스키마보내성공여부
스키마보내성공여부 = 0
PANTS_TEMP:ARG:종류 += 1
IF ITEM:콜렉션의스키마
	IF FLAG:C스키마자동사양
		CALL PANTS_SUKIMA(ARG, 종류)
		스키마보내성공여부 = RESULT
	ELSEIF FLAG:팬티수송 == 0
		PRINTFORML 콜렉션의 스키마를 사용하겠습니까? (남은 수：{ITEM:콜렉션의스키마})
		CALL ASK_YN
		IF !RESULT
			CALL PANTS_SUKIMA(ARG, 종류)
			스키마보내성공여부 = RESULT
		ENDIF
	ENDIF
ENDIF
SIF !스키마보내성공여부
	FLAG:팬티수송 = 1
RETURN 스키마보내성공여부

@PANTS_SUKIMA(ARG, 종류)
#DIM 종류
#DIM 스키마보내성공여부
스키마보내성공여부 = 0
IF PANTS_TEMP:ARG:종류
	CFLAG:ARG:(종류 + 팬티관리) += PANTS_TEMP:ARG:종류
	PANTS_TEMP:ARG:종류 = 0
	PRINTFORMW %마스터는% 콜렉션의 스키마를 열어 %CALLNAME:ARG%의 %PANTSNAME(종류, ARG)%를 던져넣었다
	ITEM:콜렉션의스키마 -= 1
	스키마보내성공여부 = 1
	;스키마 팬티 대응용. KR판에만 있음
	SIF CFLAG:ARG:(종류 + 팬티관리)
		FLAG:700 ++
ENDIF
RETURN 스키마보내성공여부

@SHOW_HAVE_COLLECTION
IF ITEM:콜렉션의스키마 == 0
	CALL DISPLAY_ALL_COLLECTION(1,0)
	PRINTW
ELSE
	PRINTFORMW 콜렉션의 스키마를 사용하겠습니까? (남은 수：{ITEM:콜렉션의스키마})
	DRAWLINE
	CALL DISPLAY_ALL_COLLECTION(1,1)
	DRAWLINE
	PRINTFORML [0] 사용하지 않는다
	INPUT
	LOCAL = RESULT
	CALL PANTS_SUKIMA(LOCAL / 100, LOCAL % 100)
	;팬티수송 대책
	SIF RESULT
		CALL SPEND_HAVE_COLLECTION(LOCAL)
ENDIF

;### 젖어 틈이 날 수 있는 파는 팬티를 입고 있어 판정용 식중함수 ################
;ARG=캐릭터 번호, 돌아가 값 1은 틈이 날 수 있는 파는, 0은 틈이 날 수 있는 없다
;루코토말로부터의 이식
@PANTIES_TRANSPARENCY(ARG)
#FUNCTION

;팬티를 빼앗은 or 받은 or 원래부터 입지 않은 경우는 0
IF CFLAG:ARG:노팬티 || (CFLAG:ARG:205 == 0 && CFLAG:ARG:206 == 0)
	RETURNF 0
;노팬티가 아닌 경우
ELSE
	;지금 현재 팬티를 입고 있는 경우
	IF EQUIP:ARG:하반신속옷1 || EQUIP:ARG:하반신속옷2
		IF EQUIP:ARG:하반신속옷1
			RETURNF GET_INT(ARG, "하반신속옷_비켜놓기불가", EQUIP:ARG:하반신속옷1, "비쳐보이는")
		ELSEIF EQUIP:ARG:하반신속옷2
			RETURNF GET_INT(ARG, "하반신속옷_비켜놓기가능", EQUIP:ARG:하반신속옷2, "비쳐보이는")
		ENDIF
	;지금 현재 입지 않지만 다시 나중에 입는 경우는 그 팬티가 틈이 날 수 있는 파는지 아닌지를 돌려준다
	ELSE
		;하반신속옷1
		IF CFLAG:ARG:205
			RETURNF GET_INT(ARG, "하반신속옷_비켜놓기불가", CFLAG:ARG:205, "비쳐보이는")
		;하반신속옷2
		ELSEIF CFLAG:ARG:206
			RETURNF GET_INT(ARG, "하반신속옷_비켜놓기가능", CFLAG:ARG:206, "비쳐보이는")
		ENDIF
	ENDIF
ENDIF

;해당 없으면 틈이 날 수 없기로 한다
RETURNF 0

@PANTS_ALREADY_COLLECTED(ARG)
#FUNCTION
#DIM 종류
SIF !ARG
	ARG = TARGET
;입고 있는 팬티의 소유 확인
IF EQUIP:ARG:하반신속옷2
	FOR 종류, 0, MAXPANTS
		IF 종류 == EQUIP:ARG:하반신속옷2
			SIF CFLAG:ARG:(종류 + 팬티관리) > 0
				RETURNF 1
		ENDIF
	NEXT
ENDIF
RETURNF 0


code:
;────────────────────────────────────
;조교 종료의 처리
;────────────────────────────────────
@EVENTEND
#LATER
#DIM 변동치
#DIM 만나지않은날짜
#DIM YANUSHI
PRINTL 하루가 끝났습니다
PRINTFORML
;SIF STRLENS(TSTR:1)
;	CALL SELL_VIDEO

;수면 플래그
CALL SET_SLEEP(1, MASTER, 0)

CFLAG:MASTER:286 = 0
CFLAG:MASTER:287 = 0
CFLAG:MASTER:288 = 0
CFLAG:MASTER:289 = 0
CFLAG:MASTER:290 = 0
CFLAG:MASTER:노팬티 = 0

;제비를 파하도록(듯이) 플래그 리셋
CLEARBIT FLAG:999, 1
;あなた懐妊処理
;CALL MasterPregnancy

;질내사정 플래그 리셋
FOR LOCAL, 1, CHARANUM
	CFLAG:LOCAL:286 = 0
	CFLAG:LOCAL:287 = 0
	CFLAG:LOCAL:288 = 0
	CFLAG:LOCAL:289 = 0
	CFLAG:LOCAL:290 = 0
	;발원함에 의한 감도 상승 해제
	SIF GROUPMATCH(TCVAR:LOCAL:발원함보정, 23,31,41,70,80,82,83,101,102,103,104,106)
		TALENT:LOCAL:(TCVAR:LOCAL:발원함보정) -= 1
	;좋아하는체위 결정
	SIF CFLAG:LOCAL:좋아하는체위 > 10
		CFLAG:LOCAL:좋아하는체위 = 0
	FOR COUNT, 1,10
		SIF EXP:LOCAL:(COUNT + 100) > EXP:LOCAL:(CFLAG:LOCAL:좋아하는체위 + 100) + 50
			CFLAG:LOCAL:좋아하는체위 = COUNT
	NEXT
	;팬티 플래그
	CFLAG:LOCAL:노팬티 = 0
	CFLAG:LOCAL:의상일시변경 = 0
	SIF TCVAR:LOCAL:본래의복장
		CFLAG:LOCAL:기본복장세트 = TCVAR:LOCAL:본래의복장
	SELECTCASE ABL:LOCAL:정액중독
		CASE 0
			LOCAL:1 = 5
		CASE 1 TO 3
			LOCAL:1 = 7
		CASE 4 TO 7
			LOCAL:1 = 10
		CASEELSE
			LOCAL:1 = 20
	ENDSELECT
	;기분
	TALENT:LOCAL:기분 = 0
	CALL KIGEN_CHANGE(LOCAL, 10)
	SIF TALENT:MASTER:농후정액
		LOCAL:1 *= 2
	SETCOLOR 0xEE60EE
	IF EX:LOCAL:질내정액 > 200 || EX:LOCAL:애널내정액 > 200
		PRINTFORM %CALLNAME:MASTER%에 충분히 쏟아진 정액으로, %CALLNAME:LOCAL%의
		IF EX:LOCAL:질내정액 > 200 && EX:LOCAL:애널내정액 > 200
			PRINTL 질내와 애널이 뜨겁게 쑤시고 있다…
		ELSEIF EX:LOCAL:질내정액 > 200
			PRINTL 질내가 뜨겁게 쑤시고 있다…
		ELSE
			PRINTL 애널이 뜨겁게 쑤시고 있다…
		ENDIF
		IF EX:LOCAL:질내정액 > 200
			JUEL:LOCAL:굴복 += LOCAL:1 * (ABL:LOCAL:질내사정중독 + 1) * EX:LOCAL:질내정액
			JUEL:LOCAL:욕정 += LOCAL:1 * (ABL:LOCAL:질내사정중독 + 1) * EX:LOCAL:질내정액
		ENDIF
		IF EX:LOCAL:애널내정액 > 200
			JUEL:LOCAL:굴복 += LOCAL:1 * (ABL:LOCAL:항내사정중독 + 1) * EX:LOCAL:애널내정액
			JUEL:LOCAL:욕정 += LOCAL:1 * (ABL:LOCAL:항내사정중독 + 1) * EX:LOCAL:애널내정액
		ENDIF
	ENDIF
	IF (EX:LOCAL:20 + EX:LOCAL:21 + EX:LOCAL:22 + EX:LOCAL:23 + EX:LOCAL:24) > 0
		PRINTFORM %CALLNAME:LOCAL%는
		SIF EX:LOCAL:20
			PRINT 클리토리스
		SIF EX:LOCAL:20 && (EX:LOCAL:21 || EX:LOCAL:22 || EX:LOCAL:23 || EX:LOCAL:24)
			PRINT 와
		SIF EX:LOCAL:21
			PRINT 질내
		SIF EX:LOCAL:21 && (EX:LOCAL:22 || EX:LOCAL:23 || EX:LOCAL:24)
			PRINT 와
		SIF EX:LOCAL:22
			PRINT 애널
		SIF EX:LOCAL:22 && (EX:LOCAL:23 || EX:LOCAL:24)
			PRINT 와
		SIF EX:LOCAL:23
			PRINT 가슴
		SIF EX:LOCAL:23 && EX:LOCAL:24
			PRINT 와
		SIF EX:LOCAL:24
			PRINT 입술
		PRINTFORMW 의 쑤심에 괴로워하고 있다…
		CFLAG:LOCAL:욕구불만도 += (EX:LOCAL:20 + EX:LOCAL:21 + EX:LOCAL:22 + EX:LOCAL:23 + EX:LOCAL:24) * 100
	ENDIF
	{
	IF (TALENT:LOCAL:150 || TALENT:LOCAL:151) && !TALENT:LOCAL:초경 && EXP:LOCAL:V절정경험 > MAX(0, 200 - CFLAG:LOCAL:자식보정)
	 && !RAND:(10 - MIN(9, CFLAG:LOCAL:자식보정 / 10 + ABL:LOCAL:질내사정중독)) && EX:LOCAL:질내정액
	 }
		PRINTFORML 어린 몸이 알기에는 너무 빠른 쾌락이 %CALLNAME:LOCAL%를 「여자」로 바꾸어 간다……
		PRINTFORMW %CALLNAME:LOCAL%는 초경을 맞이했다
		PRINTFORMW %CALLNAME:LOCAL%는 아이를 임신할 수 있는 몸이 되었다
		TALENT:LOCAL:초경 = 1
	ENDIF
	IF ABL:LOCAL:정액중독 < 3
		IF CFLAG:LOCAL:누계뿌려짐 + CFLAG:LOCAL:누계정음 * 3 +CFLAG:LOCAL:누계질내정액 + CFLAG:LOCAL:누계애널정액 >= 15000 * (10 - 2 * TALENT:LOCAL:중독되기쉬움) / 10
			PRINTFORML 몇 번이나 정액을 쏟아진 %CALLNAME:LOCAL%의 몸은 %CALLNAME:MASTER%의 정액의 맛을 기억해 버렸다…
			PRINTL 향후는 무의식적이어도 스스로 갖고 싶어하게 될 것이다
			PRINTFORMW %CALLNAME:LOCAL%의 정액중독이 3이 되었다
			ABL:LOCAL:정액중독 = 3
		ELSEIF CFLAG:LOCAL:누계뿌려짐 + CFLAG:LOCAL:누계정음 * 3 +CFLAG:LOCAL:누계질내정액 + CFLAG:LOCAL:누계애널정액 >= 7500 * (10 - 2 * TALENT:LOCAL:중독되기쉬움) / 10 && ABL:LOCAL:정액중독 < 2
			PRINTFORML %CALLNAME:LOCAL%의 몸은, 정액에 친숙해 지고 있다…
			PRINTFORMW %CALLNAME:LOCAL%의 정액중독이 2가 되었다
			ABL:LOCAL:정액중독 = 2
		ELSEIF CFLAG:LOCAL:누계뿌려짐 + CFLAG:LOCAL:누계정음 * 3 +CFLAG:LOCAL:누계질내정액 + CFLAG:LOCAL:누계애널정액 >= 3000 * (10 - 2 * TALENT:LOCAL:중독되기쉬움) / 10 && !ABL:LOCAL:정액중독
			PRINTFORML %CALLNAME:LOCAL%의 몸은, 정액의 맛을 알아 버렸다…
			PRINTFORMW %CALLNAME:LOCAL%의 정액중독이 1이 되었다
			ABL:LOCAL:정액중독 = 1
		ENDIF
	ENDIF
	IF CFLAG:LOCAL:누계정음 > 10000 * (10 - 2 * TALENT:LOCAL:중독되기쉬움) / 10 && !TALENT:LOCAL:정애미각
		PRINTFORML 정액을 계속 먹은 %CALLNAME:LOCAL%의 몸은 정액의 맛 자체에서 즐거움을 느끼게 되고 말았다…
		PRINTFORMW %CALLNAME:LOCAL%는 【정애미각】을 얻었다
		TALENT:LOCAL:정애미각 = 1
	ENDIF
	IF ABL:LOCAL:정액중독 >= 3
		IF CFLAG:LOCAL:누계질내정액 > 3000 * ABL:LOCAL:질내사정중독 * (10 - 2 * TALENT:LOCAL:중독되기쉬움) / 10 &&  EX:LOCAL:질내정액 > 200 && ABL:LOCAL:질내사정중독 < 3
			PRINTFORMW %CALLNAME:LOCAL%의 핑크색의 점막에 정액이 널리 스며든다…
			PRINTFORMW %CALLNAME:LOCAL%의 질은 \@ ABL:LOCAL:질내사정중독 ? 보다 강하게# \@%CALLNAME:MASTER%의 정액을 바라게 되었다
			ABL:LOCAL:질내사정중독 ++
		ELSEIF CFLAG:LOCAL:누계애널정액 > 3000 * ABL:LOCAL:항내사정중독 * (10 - 2 * TALENT:LOCAL:중독되기쉬움) / 10 && EX:LOCAL:애널내정액 > 200 && ABL:LOCAL:항내사정중독 < 3
			PRINTFORMW %CALLNAME:LOCAL%의 직장에 정액이 널리 스며든다…
			PRINTFORMW %CALLNAME:LOCAL%의 애널은 \@ ABL:LOCAL:항내사정중독 ? 보다 강하게# \@%CALLNAME:MASTER%의 정액을 바라게 되었다
			ABL:LOCAL:항내사정중독 ++
		ENDIF
	ENDIF
	RESETCOLOR
	IF FLAG:임신스위치 == 1
		SIF CFLAG:LOCAL:임신경과일수 == 0 && !FLAG:피임 && !TCVAR:LOCAL:피임약 && TALENT:LOCAL:스침 != 2
			CALL GET_CHILD(LOCAL)
	ELSE
	;
	ENDIF
	CALL 출산경과(LOCAL)
	SIF CFLAG:LOCAL:600
		CFLAG:LOCAL:600 = 0
	IF CFLAG:LOCAL:안면
		만나지않은날짜 = DAY - CFLAG:LOCAL:마지막에만난날
		CALL KOJO_MESSAGE_SEND("EVENT", 31, LOCAL, 만나지않은날짜)
	ENDIF
NEXT

IF CFLAG:MASTER:누계정음 > 10000 * (10 - 2 * TALENT:MASTER:중독되기쉬움) / 10 && !TALENT:MASTER:정애미각
	PRINTFORML 정액을 계속 먹은 %CALLNAME:MASTER%의 몸은 정액의 맛 자체에서 즐거움을 느끼게 되고 말았다…
	PRINTFORMW %마스터는% 【정애미각】을 얻었다
	TALENT:MASTER:정애미각 = 1
ENDIF
;自分に자식보정は掛けられないため、暫定で30回分の補正が掛った値で判定する
{
IF (TALENT:PLAYER:150 || TALENT:PLAYER:151) && !TALENT:PLAYER:초경 && EXP:PLAYER:V절정경험 > MAX(0, 200 - 30)
 && !RAND:(10 - MIN(9, 30 / 10 + ABL:PLAYER:질내사정중독)) && EX:PLAYER:질내정액
 }
	PRINTFORML 어린 몸이 알기에는 너무 빠른 쾌락이 %CALLNAME:PLAYER%를 「여자」로 바꾸어 간다……
	PRINTFORMW %CALLNAME:PLAYER%은 초경을 맞았다
	PRINTFORMW %CALLNAME:PLAYER%는 아이를 임신할 수 있는 몸이 되었다
	TALENT:PLAYER:초경 = 1
ENDIF

FOR LOCAL, MINROOM(), MAXROOM
	YOGORE:LOCAL += RAND:50
	YANUSHI = PRIVATEROOM:(LOCAL % 100)
	IF YANUSHI && YOGORE:LOCAL > (4000 - ABL:YANUSHI:청소기능 * 500)
		PRINTFORML %CALLNAME:YANUSHI%는 자신의 방을 청소했다
		EXP:YANUSHI:청소경험 ++
		YOGORE:LOCAL /= 4
	ENDIF
NEXT
IF FLAG:가호
	FLAG:가호 = 0
	DRAWLINE
	PRINTFORMW 신의 가호의 효과가 끊어졌다
ENDIF
IF FLAG:이상기상
	FLAG:이상기상 = 0
	DRAWLINE
	PRINTFORMW 이상기상은 들어갔다
ENDIF
IF FLAG:패션
	FLAG:패션 = 0
	DRAWLINE
	PRINTFORMW 패션게임이 진정되었다
ENDIF
IF GatheringSeason
	GatheringSeason = 0
	PRINTFORMW 계절이 제자리를 찾았다
ENDIF
IF TFLAG:출장청소
	DRAWLINE
	PRINTFORML 부탁해둔 청소 서비스가 왔다
	PRINTFORMW …
	PRINTFORMW ……
	PRINTFORMW %GET_MAPNAME(MAIN_MAP)% 안이 깨끗해졌다!
	FOR LOCAL, MINROOM(), MAXROOM
		YOGORE:LOCAL /= 4
	NEXT
ENDIF

[SKIPSTART]
;의뢰의 기한 마감을 체크
FOR LOCAL, 1,6
	SIF !IRAI_SLOT:LOCAL
		CONTINUE
	SIF IRAI_DEADLINE:LOCAL > DAY
		CONTINUE
	DRAWLINE
	PRINTFORML %CALLNAME:(IRAI_SLOT_TO_CLIENT(LOCAL))%로부터의 의뢰의 기한이 지났다……
	CALL COLORMESSAGE("【의뢰 실패】", C_YELLOW, 2,1)
	CFLAG:(IRAI_CLIENT:LOCAL):의뢰상황 = 4
NEXT
[SKIPEND]

;데일리
;SIF CFLAG:MASTER:현재위치 != OMANEKIBEYA() && CFLAG:MASTER:데이트중 == MAIN_MAP
SIF CFLAG:MASTER:현재위치 == CFLAG:MASTER:초기위치
	CALL DAIRYEVENT
;밭의 관리
CALL YASAI

;대본소
CALL BOOKLEND

CALL 주조

CALL MOENY_HENDOU

MONEY:1 += MAXBASE:MASTER:TSP - BASE:MASTER:TSP
IF MAXBASE:MASTER:TSP >= 4000&&(MAXBASE:MASTER:TSP - BASE:MASTER:TSP)>MAXBASE:MASTER:TSP/2 ; 첫 IF행 KR판 구잔재 또는 오리지날
	DRAWLINE
	PRINTFORML TSP가 {(MAXBASE:MASTER:TSP - BASE:MASTER:TSP)/100}만큼 올랐다.
	MAXBASE:MASTER:TSP+=(MAXBASE:MASTER:TSP - BASE:MASTER:TSP)/100
ELSEIF MONEY:1 >= 30000 && MAXBASE:MASTER:TSP == 2500
	DRAWLINE
	PRINTFORML 시간정지 능력이 극한 진화했다!
	MAXBASE:MASTER:TSP = 4000
ELSEIF MONEY:1 >= 10000 && MAXBASE:MASTER:TSP == 1500
	DRAWLINE
	PRINTFORML 시간정지 능력의 취급이 최고로 능숙해졌다!
	MAXBASE:MASTER:TSP = 2500
ELSEIF MONEY:1 >= 5000 && MAXBASE:MASTER:TSP == 800
	DRAWLINE
	PRINTFORML 시간정지 능력의 취급이 꽤 능숙해졌다!
	MAXBASE:MASTER:TSP = 1500
ELSEIF MONEY:1 >= 1800 && MAXBASE:MASTER:TSP == 500
	DRAWLINE
	PRINTFORML 시간정지 능력의 취급이 능숙해졌다!
	MAXBASE:MASTER:TSP = 800
ELSEIF MONEY:1 >= 500 && MAXBASE:MASTER:TSP == 200
	DRAWLINE
	PRINTFORML 시간정지 능력의 취급이 조금 능숙해졌다!
	MAXBASE:MASTER:TSP = 500
ELSEIF MONEY:1 >= 150 && MAXBASE:MASTER:TSP == 100
	DRAWLINE
	PRINTFORML 시간정지 능력의 취급이 살짝 능숙해졌다!
	MAXBASE:MASTER:TSP = 200
ENDIF

BASE:MASTER:TSP = MAXBASE:MASTER:TSP

IF TALENT:MASTER:2 & 2
	IF BASE:MASTER:정력 == MAXBASE:MASTER:정력 && !GETBIT(TFLAG:일일일회, 8)
		FLAG:71 ++
		IF FLAG:71 >= 5 && !RAND:3
			DRAWLINE
			PRINTFORML 최근 뽑지 않은 키 있고로 응 짜증 한다……!
			PRINTFORML %마스터는% 이불 속에서 몸부림쳐 돌았다
			CALL COLORMESSAGE("스트레스로 일시적으로 기력의 최대치가 감소했다", C_YELLOW, 2,1)
			변동치 = MIN(MAXBASE:MASTER:기력 * 20 / 100, 500)
			CALL BUFF_BASE(MASTER, BASE_기력,-변동치)
		ELSEIF FLAG:71 >= 3
			TALENT:MASTER:120 = 1
			PRINTL 그러고 보니  요즈음 뽑지 않았다…
		ENDIF
	ELSE
		FLAG:71 = 0
		TALENT:MASTER:120 = 0
	ENDIF
	IF BASE:MASTER:정력 < MAXBASE:MASTER:정력 * 2 / 10 && BASE:MASTER:체력 > 500
		DRAWLINE
		PRINTFORML 오늘은 가득 사정하고 싶다……
		PRINTFORML %마스터는% 기분 좋은 권태감 속에서 행복하게 잠들었다
		CALL COLORMESSAGE("일시적으로 TSP와 기력의 최대치가 증가했다", C_YELLOW, 2,1)
		변동치 = MIN(MAXBASE:MASTER:기력 / 10, 300)
		CALL BUFF_BASE(MASTER, BASE_기력, 변동치)
		변동치 = MIN(MAXBASE:MASTER:TSP / 2, 500)
		CALL BUFF_BASE(MASTER, BASE_TSP, 변동치)
	ENDIF
ENDIF

CALL ONANISM

CALL GIFT_FAVORITE

SIF FLAG:술집매입강화 > 0
	FLAG:술집매입강화 --
;무슨주를 얻을 수 있었는지
FOR LOCAL, 0, CHARANUM
	SIF CFLAG:LOCAL:출금
		CONTINUE
	CALL SOURCE_ABLUP(LOCAL)
	CALL JUEL_CHECK(LOCAL)
	CALL DENIAL_CHECK(LOCAL)
	SIF TALENT:LOCAL:스침
		CFLAG:LOCAL:출금 = 2
NEXT

BEGIN ABLUP

@JUEL_CHECK(ARG)
FOR LOCAL, 0,100
	IF PALAM:ARG:LOCAL < PALAMLV:1
		LOCAL:1 = 0
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:1*3
		LOCAL:1 = 1
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:2
		LOCAL:1 = 2
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:2*3
		LOCAL:1 = 10
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:3
		LOCAL:1 = 20
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:3*2
		LOCAL:1 = 100
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:4
		LOCAL:1 = 200
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:5
		LOCAL:1 = 1000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:6
		LOCAL:1 = 2000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:7
		LOCAL:1 = 3000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:8
		LOCAL:1 = 5000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:9
		LOCAL:1 = 8000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:10
		LOCAL:1 = 12000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:11
		LOCAL:1 = 25000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:12
		LOCAL:1 = 40000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:13
		LOCAL:1 = 60000
	ELSEIF PALAM:ARG:LOCAL < PALAMLV:14
		LOCAL:1 = 100000
	ELSE
		LOCAL:1 = 150000
	ENDIF
	;쾌감계
	IF LOCAL >= 0 && LOCAL < 9
		GOTJUEL:ARG:LOCAL = LOCAL:1 + EX:ARG:LOCAL * 1000
	;그 외
	ELSEIF LOCAL < 30
		GOTJUEL:ARG:LOCAL = LOCAL:1
	;부정의 구슬
	ELSEIF LOCAL == 30 || LOCAL == 31 || LOCAL == 32
		GOTJUEL:ARG:100 += LOCAL:1
	ENDIF
NEXT
JUEL:ARG:100 += GOTJUEL:ARG:100
FOR LOCAL, 0,100
	SIF !STRLENS(PALAMNAME:LOCAL)
		CONTINUE
	SIF LOCAL != 9 &&LOCAL != 30 && LOCAL != 31 && LOCAL != 32
		JUEL:ARG:LOCAL += GOTJUEL:ARG:LOCAL
NEXT
@DENIAL_CHECK(ARG)
SIF JUEL:ARG:100 == 0
	RETURN 0
LOCAL = 1
WHILE LOCAL
	SIF !JUEL:ARG:부정 || (!JUEL:ARG:복종 && !JUEL:ARG:욕정 && !JUEL:ARG:굴복)
		BREAK
	LOCAL:1 = JUEL:ARG:부정 == 1 ? 1 # JUEL:ARG:부정 / 2
	;JUEL:10~12로 소비
	LOCAL:2 = RAND:3 + 10
	SIF LOCAL:1 > JUEL:ARG:(LOCAL:2)
		LOCAL:1 = JUEL:ARG:(LOCAL:2)
	JUEL:ARG:부정 -= LOCAL:1
	JUEL:ARG:(LOCAL:2) -= LOCAL:1
WEND
LOCAL = 1
WHILE LOCAL
	SIF !JUEL:ARG:부정 || (!JUEL:ARG:치정 && !JUEL:ARG:고통 && !JUEL:ARG:공포 && !JUEL:ARG:호의 && !JUEL:ARG:우월)
		BREAK
	LOCAL:1 = JUEL:ARG:부정 == 1 ? 1 # JUEL:ARG:부정 / 2
	;JUEL:14~16으로 소비
	LOCAL:2 = RAND:5
	SELECTCASE LOCAL:2
		CASE 0
			LOCAL:2 = 14
		CASE 1
			LOCAL:2 = 15
		CASE 2
			LOCAL:2 = 16
		CASE 3
			LOCAL:2 = 20
		CASE 4
			LOCAL:2 = 21
	ENDSELECT
	SIF LOCAL:1 > JUEL:ARG:(LOCAL:2)
		LOCAL:1 = JUEL:ARG:(LOCAL:2)
	JUEL:ARG:부정 -= LOCAL:1
	JUEL:ARG:(LOCAL:2) -= LOCAL:1
WEND

@ONANISM
#DIM 욕구불만
#DIM 만족도
#DIM 질투도
#DIM 구슬입수배율
#DIM 부위지정
#DIM 추가V자위
#DIM 추가A자위
#DIM 추가B자위
#DIM 자위타입
#DIM 자위발생최소치
#DIM CONST 문자사아즈 = 15
#DIM 자위화상
VARSET LOCAL

FOR LOCAL,1,CHARANUM
	SIF CFLAG:LOCAL:출금
		CONTINUE
	부위지정 = 0
	자위타입 = 0
	추가V자위 = 0
	추가A자위 = 0
	추가B자위 = 0
	자위발생최소치 = LIMIT(750 - 200 * TALENT:LOCAL:자위하기쉬움 - 100 * TALENT:LOCAL:쾌감응답 - 100 * TALENT:LOCAL:자기애 + 200 * TALENT:LOCAL:자제심, 500,999)
	자위발생최소치 -= 300 * TALENT:LOCAL:자위중독
	IF CFLAG:LOCAL:딸딸이들킴
		PRINTFORML %CALLNAME:LOCAL%는 베개에 얼굴을 묻어 푸드득푸드득 하고 있다…
	ELSEIF CFLAG:LOCAL:욕구불만도 > 자위발생최소치
		SIF !FLAG:自慰画像非表示
			CALL PrintOnanismImage(LOCAL)
		LOCAL:1 = 0
		IF TALENT:LOCAL:애욕
			LOCAL:1 = 4
		ELSEIF TALENT:LOCAL:연모
			LOCAL:1 = 3
		ELSEIF TALENT:LOCAL:섹프
			;정부도 여기
			LOCAL:1 = 2
		ELSEIF TALENT:LOCAL:사모
			LOCAL:1 = 1
		ENDIF
		IF TALENT:LOCAL:자제심 && LOCAL:1 < 5
			LOCAL:1 += 1
		ENDIF
		IF CFLAG:LOCAL:기정사실 & 합의_우후후 && !RAND:(8 - LOCAL:1)
			PRINTFORML %CALLNAME:LOCAL%는 몸의 쑤심에 괴로워하면서도, 내일 %CALLNAME:MASTER%에 안아주려고 생각해 당신을 억지로 잠에 올랐다……
			자위타입 = 7
			;CALL KOJO_MESSAGE_SEND("EVENT", 22, LOCAL, 7)
			CFLAG:LOCAL:욕구불만도 = LIMIT(CFLAG:LOCAL:욕구불만도 + 100, 0 , 1000)
			SIF !CFLAG:LOCAL:신사거주
				CFLAG:LOCAL:다음날내방 = 1
			PRINTL
		ELSEIF CFLAG:MASTER:현재위치 == CFLAG:LOCAL:초기위치
			PRINTFORML 한밤중, %마스터는% 소리를 들어 눈을 떴다
			PRINTFORML ……%CALLNAME:LOCAL%가 %CALLNAME:MASTER%의 손을 잡으면서 소리를 눌러 참아 자위에 빠지고 있었다……
			자위타입 = 2
		ELSEIF ABL:LOCAL:정액중독 > 2 && EX:LOCAL:질내정액 && !RAND:4
			PRINTFORML 참을 수 없어진 %CALLNAME:LOCAL%는, 질내에 남는 %CALLNAME:MASTER%의 정액을 박아 넣어 친숙해 지게 하도록(듯이) 자위를 했다……
			자위타입 = 3
		ELSEIF TALENT:LOCAL:무지
			LOCAL:1 = 0
			LOCAL:1 = EXP:LOCAL:자위경험 - EXP:LOCAL:조교자위경험
			IF LOCAL:1 >= 120 - (MIN(ABL:LOCAL:친밀, 6) * 10) && !FLAG:무지자동상실금지
				PRINTFORMW 마침내 %CALLNAME:LOCAL%는 보호자에게 상담했다……
				PRINTFORMW %CALLNAME:LOCAL%는 무지를 잃었다……
				TALENT:LOCAL:무지 = 0
			ELSEIF LOCAL:1 >= 80 - (MIN(ABL:LOCAL:친밀,4) * 10)
				PRINTFORML %CALLNAME:LOCAL%는 자위를 자위와 모르는 채 자위에 몰두하고 있다……
			ELSEIF LOCAL:1 >= 50 - (MIN(ABL:LOCAL:친밀,3) * 10)
				PRINTFORML %CALLNAME:LOCAL%는 책상의 모퉁이에 가랑이를 꽉 누르면서 허리를 거절해, 비소를 자극했다……
			ELSEIF LOCAL:1 >= 30 - (MIN(ABL:LOCAL:친밀,2) * 10)
				PRINTFORML %CALLNAME:LOCAL%는 책을 가랑이에 끼워 어린 비렬을 자극했다……
			ELSEIF LOCAL:1 >= 15 - (MIN(ABL:LOCAL:친밀,1) * 10)
				PRINTFORML %CALLNAME:LOCAL%는 공부중에 붓을 고간에 맞혀, 말랑말랑의 균열부를 켰다 ……
			ELSE
				PRINTFORML %CALLNAME:LOCAL%는 침상에 옆으로 누워서면서 태퇴를 조정 괴로워하고 있다……
			ENDIF
			자위타입 = 4
		ELSEIF ABL:LOCAL:자위중독 > 3 && (TALENT:LOCAL:사모 || TALENT:LOCAL:연모 || TALENT:LOCAL:연인 || TALENT:LOCAL:애욕 || TALENT:LOCAL:섹프) && !RAND:3
			PRINTFORML 어떻게도 참을 수 없게 된%CALLNAME:LOCAL%는, 옷에 배어든%CALLNAME:MASTER%의 냄새를 맡아면서 격렬한 자위를 했다……
			LOCAL:2 = RAND:5
			IF (TALENT:LOCAL:C감도 == 1 || ABL:LOCAL:C감각 >= 10) && LOCAL:2 == 4
				PRINTFORM 음핵을 깎아, 애액을 분출하면서 지관, 행위에 몰두했지만
				부위지정 = 1
			ELSEIF (TALENT:LOCAL:V감도 == 1 || ABL:LOCAL:V감각 >= 10) && LOCAL:2 == 3
				PRINTFORM 꿀 단지에 손가락을 깊숙이 삽입해 애액을 계속 긁어냈지만
				부위지정 = 2
				추가V자위 = 1
			ELSEIF (TALENT:LOCAL:A감도 == 1 || ABL:LOCAL:A감각 >= 10) && LOCAL:2 == 2
				PRINTFORM 비소 뿐만이 아니라 항문도 요계속했지만
				부위지정 = 3
				추가A자위 = 1
			ELSEIF (TALENT:LOCAL:B감도 == 1 || ABL:LOCAL:B감각 >= 10) && LOCAL:2 == 1
				IF TALENT:LOCAL:가슴사이즈 >= 1
					PRINTFORM 스스로의 풍만한 유방을 질네라고 빨아 올려
				ELSEIF TALENT:LOCAL:가슴사이즈 <= -1
					PRINTFORM 얇은 가슴을 어루만지고 돌리면서 유두를 문지르며,
				ELSE
					PRINTFORM 부드러운 유방을 부드럽게 비비어 하기 만여,
				ENDIF
				PRINTFORM 몇 번이고 절정을 맞이하는 것도
				부위지정 = 4
				추가B자위 = 1
			ELSE
				;PRINTFORML 그러나 아무리 쾌감을 얻으려 해도 %CALLNAME:MASTER%에의 구상은 다스려질 것은 없고, %CALLNAME:LOCAL%는 아침까지 괴로워한 시간을 보냈다……
				PRINTFORM 체내에 울리는 열락에 몸을 진동시키면서도
				부위지정 = 5
			ENDIF
			IF TALENT:LOCAL:사모 || TALENT:LOCAL:연모 || TALENT:LOCAL:연인
				PRINTFORML , %CALLNAME:MASTER%에의 구상은 다스려질 것은 없었다……
			ELSE
				PRINTFORML , 타오른 몸의 열은 들어가지 않았다……
			ENDIF
			추가V자위 = 1
			자위타입 = 5
		ELSEIF ABL:LOCAL:자위중독 > 3 && !RAND:2
			PRINTFORM 이미 손가락에서는 만족할 수 없어진 %CALLNAME:LOCAL%는 캇파로부터 입수한 완구
			IF ABL:LOCAL:A감각 >= 5
				IF TALENT:LOCAL:정조 == -1 && TALENT:LOCAL:처녀 == 0
					PRINTFORM 를 질과 아누스의 깊숙한 곳까지 질네 붐비어
					추가V자위 = 1
				ELSEIF TALENT:LOCAL:정조 == 1
					PRINTFORM 를 비소와 항문에 꽉 눌러
				ELSE
					PRINTFORM 로 비소와 항문을 만지작거려
				ENDIF
				부위지정 = 3
				추가A자위 = 1
			ELSE
				IF TALENT:LOCAL:정조 == -1 && TALENT:LOCAL:처녀 == 0
					PRINTFORM 를 질의 깊숙한 곳까지 질네 붐비어
					추가V자위 = 1
				ELSEIF TALENT:LOCAL:정조 == 1
					PRINTFORM 를 비소에 꽉 눌러
				ELSE
					PRINTFORM 로 비소를 만지작거려
				ENDIF
				부위지정 = 1
			ENDIF
			PRINTFORML 자위에 빠지고 있다…
			자위타입 = 6
		ELSE
			PRINTFORML 쑤시는 몸을 끝내 억누를 수 없어진 %CALLNAME:LOCAL%는 음부를 자기 손가락으로 만지작거렸다…
			자위타입 = 1
		ENDIF
		CALL KOJO_MESSAGE_SEND("EVENT", 22, LOCAL, 자위타입, 부위지정)
		SIF 자위타입 == 7
			CONTINUE
		EXP:LOCAL:자위경험 += ABL:LOCAL:자위중독 * 2 + 1
		CALL COLORMESSAGE(@"자위경험＋{ABL:LOCAL:자위중독 * 2 + 1}", C_PINK, 1)
		SELECTCASE ABL:LOCAL:자위중독
			CASE 0 TO 2
				구슬입수배율 = 5
			CASE 3 TO 5
				구슬입수배율 = 3
			CASEELSE
				구슬입수배율 = 1
		ENDSELECT
		;쾌C
		CALL JUEL_UP_SPECIAL(LOCAL,"쾌C","C감각", 구슬입수배율, 1)
		SIF 추가V자위
			CALL JUEL_UP_SPECIAL(LOCAL,"쾌V","V감각", 구슬입수배율, 1)
		SIF 추가A자위
			CALL JUEL_UP_SPECIAL(LOCAL,"쾌A","A감각", 구슬입수배율, 1)
		SIF CFLAG:LOCAL:패배자위 == 1
			CALL JUEL_UP_SPECIAL(LOCAL,"고통","마조끼", 구슬입수배율 * ABL:LOCAL:마조끼, 1)
		;치정
		CALL JUEL_UP_SPECIAL(LOCAL,"치정","자위중독", 구슬입수배율 * 2,1)
		;욕정
		CALL JUEL_UP_SPECIAL(LOCAL,"욕정","욕망", 구슬입수배율, 1)
		CFLAG:LOCAL:욕구불만도 /= 4
		PRINTW
	ENDIF
	;우선 욕망으로 기초치를 결정한다
	SELECTCASE ABL:LOCAL:욕망
		CASE 0 TO 2
			욕구불만 = 10
		CASE 3 TO 5
			욕구불만 = 20
		CASE 6 TO 10
			욕구불만 = 30
		CASE 10 TO 15
			욕구불만 = 50
		CASEELSE
			욕구불만 = 80
	ENDSELECT

	욕구불만 += ABL:LOCAL:자위중독 * 7 + (ABL:LOCAL:질내사정중독 + ABL:LOCAL:항내사정중독) * 3
	욕구불만 = 욕구불만소질보정(LOCAL, 욕구불만)
	SELECTCASE 욕구불만
		CASE IS > 400
			욕구불만 = (욕구불만 - 400) / 10 + 400
		CASE IS > 200
			욕구불만 = (욕구불만 - 200) / 4 + 200
		CASE IS > 100
			욕구불만 = (욕구불만 - 100) / 2 + 100
	ENDSELECT
	SIF 자위타입
		욕구불만 = 0
	SIF TCVAR:LOCAL:오늘만났는지
		욕구불만 /= 2
	SIF TALENT:LOCAL:임신
		욕구불만 /= 5
	SIF TALENT:LOCAL:육아중
		욕구불만 /= 2
	;욕구불만도는0~1000의 사이에
	CFLAG:LOCAL:욕구불만도 = LIMIT(CFLAG:LOCAL:욕구불만도 + 욕구불만, 0, 1000)
	DEBUGPRINTFORML %CALLNAME:LOCAL%의 만족도：{만족도}　욕구 불만 상승치：{욕구불만}　욕구불만도：{CFLAG:LOCAL:욕구불만도}
NEXT

@DAIRYEVENT
;우선 30만.
#DIM EVENT_DAY, 30

CALL FISHER_YATES_SHAFFLE(30)
ARRAYCOPY "SHAFFLE_ARRAY" , "EVENT_DAY"

DRAWLINE
FOR LOCAL, 0,30
	RESULT = 0
	TRYCALLFORM DAIRY_CONDITION_EV{EVENT_DAY:LOCAL}
	IF RESULT > 0
		DEBUGPRINTFORML 데일리 이벤트{EVENT_DAY:LOCAL}가 발생했어
		DRAWLINE
		BREAK
	ENDIF
	SIF RESULT == -1
		DEBUGPRINTFORML 데일리 이벤트{EVENT_DAY:LOCAL}는 발생 조건을 채우지 않았다
NEXT
;PRINTFORMW
PRINTFORML

@욕구불만소질보정(ARG, ARG:1)
#FUNCTION
LOCAL = ARG:1
SIF TALENT:ARG:자제심
	LOCAL /= 2
SIF TALENT:ARG:자기애 < 0
	LOCAL /= 2
SIF TALENT:ARG:자기애 > 0
	LOCAL = LOCAL * 5 / 4
SIF TALENT:ARG:유치 && !TALENT:ARG:초경
	LOCAL /= 4
SIF TALENT:ARG:임신
	LOCAL /= 4
SIF TALENT:ARG:육아중
	LOCAL /= 2
SIF CFLAG:ARG:딸딸이들킴
	LOCAL /= 4
RETURNF LOCAL

license: mit


code:
@AFTER_AFFAIR
#DIM SPENT_TIME
SPENT_TIME = (TIME-UFUFU_START_TIME) / 60
IF SPENT_TIME < 0
	SPENT_TIME += 24
ENDIF
;ブチギレと無理矢理状態とTSP切れと泥酔姦と風呂場プレイと睡眠時間中は表示しない
SIF CFLAG:빡침 || TCVAR:억지로 || !SHIRAHU(TARGET)
	RETURN 0
	;ほんのちょっとのうふふでは表示されないよう、潤滑、欲情、体力、溜まってる度、その日の絶頂回数に条件をつける（大した意味はない）
SIF !(PALAM:윤활 >= PALAMLV:3 && PALAM:욕정 >= PALAMLV:3 && BASE:체력 < MAXBASE:체력 * 9 / 10 && CFLAG:욕구불만도 < 300 && EX:C절정 + EX:V절정 + EX:A절정 + EX:B절정 + EX:M절정 >= 4)
	RETURN 0

PRINTFORMD %마스터와% %CALLNAME:TARGET%\@ GET_TARGETNUM() > 1? 들은, # %조사만처리(CALLNAME:TARGET,"는")%, \@
;経過時間表示
IF SPENT_TIME
	PRINTFORMD {SPENT_TIME} 시간에 이르는
	SIF SPENT_TIME > 5
		PRINTFORMD 장렬한
ENDIF
;ここまで
IF TALENT:연모 || TALENT:연인
	PRINTFORMD 애정으로 가득 찬 인연을
ELSEIF TALENT:애욕
	PRINTFORMD 육욕으로 되는대로 서로 탐닉하는 듯한 성관계를
ELSEIF TALENT:사모
	PRINTFORMD 서로의 관계를 깊게 만드는 밀회를
ELSEIF TALENT:섹프 == 1
	PRINTFORMD 쾌락만을 목적으로 한 성행위를
ELSE
	PRINTFORMD 기세에 맡긴 경솔한 정사를
ENDIF
PRINTFORMW 끝냈다.

SIF BATHROOM()
	RETURN 0

;-------------------------------------------------------------------------------------
;以下、精液漏れ・愛液による濡れの描写（精液漏れの描写が優先）

;膣충전율が水準以上もしくはアナル충전율が水準以上
IF SOAKED_OR_NOT("V_SEMINAL") || SOAKED_OR_NOT("A_SEMINAL")
	PRINTFORMD 대량으로 %TEXTR("쏟아부어진/풀어진/주입된")% %조사처리(EXPR_VAR("SPERM"),"가")%, %CALLNAME:TARGET%의
	SIF SOAKED_OR_NOT("V_SEMINAL")
		PRINTFORMD %EXPR_VAR("VULVA")%
	SIF SOAKED_OR_NOT("A_SEMINAL")
		PRINTFORMD \@ SOAKED_OR_NOT("V_SEMINAL") ? %조사만처리(EXPR_VAR("VULVA"),"와")%  # \@%EXPR_VAR("ANUS")%
	PRINTFORMDL %TEXTR("에서 넘치려 하고 있다/에서 새어 나오고 있다/에서 흘러나오려 하고있다")%.
	IF TALENT:MASTER:농후정액
		PRINTFORML 그래도 반고형의 %조사처리(EXPR_VAR("SPERM"),"는")% 떨어지지 않고 끈 모양으로 늘어져, %CALLNAME:TARGET%의 허벅지에 달라붙었다.
	ELSE
		PRINTFORMD %타겟이% %TEXTR("화장지로/손수건으로/수건으로")%
		PRINTFORMDL %조사처리(EXPR_VAR("CROTCH"),"를")% 닦아도, 잠시 후 %조사처리(EXPR_VAR("SPERM"),"가")% %TEXTR("똑똑 늘어지며 떨어졌다/방울져 떨어지기 시작했다/흘러내려서, 허벅지에 쭉하고 선을 그었다/찔끔찔끔 넘쳐 나왔다")%.
	ENDIF
	;--------------------------------------------
	;垂れ口上呼び出し
	;--------------------------------------------
	CALL WEAREDPANTIES_TRANSPARENCY("SPERM")
	CALL KOJO_MESSAGE_SEND("EVENT",15,TARGET,3,1)
;膣・アナル충전율がともに水準未満だが潤滑が必要LV以上
ELSEIF SOAKED_OR_NOT("SECRETIONAL")
	PRINTFORMD %타겟은%、흠뻑 젖은 %조사처리(EXPR_VAR("CROTCH"),"를")% %TEXTR("화장지로/손수건으로/수건으로")%
	PRINTFORMDL 닦았지만, 흥분이 완전히 식지 않은 육체에서는 아직 꿀이 넘쳐나는 것 같다.
	;--------------------------------------------
	;溢れ口上呼び出し
	;--------------------------------------------
	CALL WEAREDPANTIES_TRANSPARENCY("JUICE")
	CALL KOJO_MESSAGE_SEND("EVENT",15,TARGET,3,2)
ENDIF
;-------------------------------------------------------------------------------------------------------
;以下、事後のムードの描写
;複数TARGETがいる場合非表示
IF GET_TARGETNUM() == 1
	PRINTFORMD 정사의 뒤처리를 끝낸 두 사람
	IF TALENT:연모 || TALENT:연인
		PRINTFORMDL 였지만, 여전히 정담을 주고받았다.
	ELSEIF TALENT:애욕
		PRINTFORMDL 였지만, 성관계의 격렬함 때문에, 좀 더 숨을 고를 시간이 필요했다.
	ELSEIF TALENT:사모
		PRINTFORMDL 은, 약간 부끄럽고, 서로 수줍어했다.
	ELSEIF TALENT:섹프 == 1
		PRINTFORMDL 은, 나른한 분위기 속에서, 잠시 말없이 서로 몸을 맞댔다.
	ELSE
		PRINTFORMDL  사이에, 뭐라 말할 수 없이 어색한 공기가 흘렀다.
	ENDIF
	PRINTFORMW
ENDIF

CALL KOJO_MESSAGE_SEND("EVENT",15,TARGET,3,3)

;---------------------------------------------------------------------------------------------------------
;以下、事後に服等を着なおす描写
;上半身上着、ワンピース、着物、外衣のどれかを着たままだったばあい
IF EQUIP:상반신윗도리1 || EQUIP:상반신윗도리2 || EQUIP:원피스 || EQUIP:기모노 || EQUIP:외투
	PRINTFORMDL 옷을 입은 채로 있던 %타겟은%, 이쪽을 등지고 흐트러진 옷을 정돈하고 있다.
	IF PALAM:욕정 >= PALAMLV:7
		PRINTFORMD 땀을 흘리고 하얀 피부가 분홍색으로 달아오른
	ELSE
		PRINTFORMD 땀을 흘린
	ENDIF
	PRINTFORMW 목덜미가 선정적이다.
;現在上半身が裸でかつ普段着の상반신속옷がブラジャーかマイクロブラのばあい
ELSEIF TEQUIP:상반신착의상황 == 0 && (CFLAG:TARGET:207 == CLOTHES_ID("상반신속옷1", "브래지어") || CFLAG:TARGET:208 == CLOTHES_ID("상반신속옷2", "마이크로브라"))
	PRINTFORMDL 정사를 마친 %타겟은%, %마스터를% 등지고, 브래지어의 후크를 끼우고 있다.
	PRINTFORMW  뒷짐진 등에 도드라지는 견갑골이 아름답다.
ENDIF

CALL KOJO_MESSAGE_SEND("EVENT",15,TARGET,3,4)

;----------------------------------------------------------------------------------------------------------
;以下、残り香の描写

;濃厚な残り香の表示に必要な射精数、噴乳数、放尿数の設定
LOCAL:5 = 5 ;必要射精数
LOCAL:6 = 5 ;必要噴乳数
LOCAL:7 = 1 ;必要放尿数

;残り香を表示するのは屋内のときだけ
IF OPENPLACE(CFLAG:MASTER:현재위치) < 2
	;その日のふたりの射精数、噴乳数、放尿数によって分岐
	IF EX:사정 + EX:MASTER:사정 >= LOCAL:5 || EX:분유 + EX:MASTER:분유 >= LOCAL:6 || EX:방뇨 + EX:MASTER:방뇨 >= LOCAL:7
		PRINTFORMD 뭉클한 열기 속에,
		SIF EX:사정 + EX:MASTER:사정 >= LOCAL:5
			PRINTFORMD 정액의 독특한 냄새
		SIF EX:분유 + EX:MASTER:분유 >= LOCAL:6
			PRINTFORMD \@ EX:사정 + EX:MASTER:사정 >= LOCAL:5 ? 와  # \@달콤한 모유 향기
		SIF EX:방뇨 + EX:MASTER:방뇨 >= LOCAL:7
			PRINTFORMD \@ EX:사정 + EX:MASTER:사정 >= LOCAL:5 || EX:분유 + EX:MASTER:분유 >= LOCAL:6 ? 와  # \@소변의 악취
		PRINTFORMDL 가 섞인 음미한 잔향이 농후하게 감돌고,
		PRINTFORMW 그 곳에서 행해진 비밀스러운 행위를 분명하게 말해준다.
	ELSE
		PRINTFORMW  새콤달콤한 체액 냄새와 뜨거운 공기가 방금까지의 정사 흔적을 남기고 있다.
	ENDIF
ENDIF
;----------------------------------------------------------------------------------------------------------

@WEAREDPANTIES_TRANSPARENCY(ARGS)
#DIM 팬티상황

;パンツを奪ったorもらったor元から穿いてない場合
IF CFLAG:노팬티 || (CFLAG:205 == 0 && CFLAG:206 == 0)
	팬티상황 = 0
;パンツを一旦脱いでいた場合
ELSEIF !EQUIP:하반신속옷1 && !EQUIP:하반신속옷2
	팬티상황 = 1
ELSE
	팬티상황 = 2
ENDIF

IF !팬티상황
	PRINTFORMDL %CALLNAME:TARGET%에게는 입을 팬티가 없다.
	IF ARGS == "SPERM"
		PRINTFORMDL 오늘은, 노팬티로 %조사처리(EXPR_VAR("SPERM"),"를")% 늘어뜨린 채 지내게 될 것 같다.
	ELSE
		PRINTFORMDL 오늘은, %조사처리(EXPR_VAR("CROTCH"),"를")% 촉촉히 적신 후에 노팬티로 지내게 될 것 같다.
	ENDIF
;パンツがある場合
ELSE
	SIF 팬티상황 == 1
		PRINTFORMDL %타겟은% 벗었던 속옷을 다시 입었다.
	IF ARGS == "SPERM"
		PRINTFORMD 흘러나온 %조사처리(EXPR_VAR("SPERM"),"가")%
	ELSE
		PRINTFORMD 끝없이 흘러나오는 애액이
	ENDIF
	PRINTFORMD \@ 팬티상황 == 1?  그# \@ 속옷을 %TEXTR("질척질척하게/흠뻑 축축하게/흠뻑 적시게")%
	;パンツが透けうる場合
	IF PANTIES_TRANSPARENCY(TARGET)
		PRINTFORMDL  해서, 비렬의 형태가 훤히 들여다보일 정도였다.
	ELSE
		PRINTFORMDL  했다.
	ENDIF
ENDIF

;### びしょ濡れ判定用式中関数 ################
;ARG=キャラ番号
@SOAKED_OR_NOT(ARGS,ARG)
#FUNCTION
#DIM 필요질충전율
#DIM 필요애널충전율
#DIM 필요윤활레벨

SIF ARG == 0
	ARG = TARGET

필요질충전율 = 30
필요애널충전율 = 30
필요윤활레벨 = PALAMLV:5

IF ARGS == "V_SEMINAL"
	IF 충전율(ARG,1) >= 필요질충전율
		RETURNF 1
	ELSE
		RETURNF 0
	ENDIF
ELSEIF ARGS == "A_SEMINAL"
	IF 충전율(ARG,2) >= 필요애널충전율
		RETURNF 1
	ELSE
		RETURNF 0
	ENDIF
ELSEIF ARGS == "SECRETIONAL"
	IF PALAM:ARG:윤활 >= 필요윤활레벨
		RETURNF 1
	ELSE
		RETURNF 0
	ENDIF
ENDIF

THROW 부정한 문자열인수

;### 表現バリエーション用式中関数 ################
@EXPR_VAR(ARGS)
#FUNCTIONS
VARSET LOCALS

IF ARGS == "SPERM"
	SELECTCASE RAND:4
	CASE 0
		LOCALS = 백탁액
	CASE 1
		LOCALS = 정액
	CASE 2
		LOCALS = 정액
	CASEELSE
		LOCALS = 정자
	ENDSELECT
ELSEIF ARGS == "VULVA"
	SELECTCASE RAND:4
	CASE 0
		LOCALS = 은밀한 곳
	CASE 1
		LOCALS = 비렬
	CASE 2
		LOCALS = 음문
	CASEELSE
		LOCALS = 여성의 음부
	ENDSELECT
ELSEIF ARGS == "ANUS"
	SELECTCASE RAND:3
	CASE 0
		LOCALS = 엉덩이의 구멍
	CASE 1
		LOCALS = 항문
	CASEELSE
		LOCALS = 아누스
	ENDSELECT
ELSEIF ARGS == "CROTCH"
	SELECTCASE RAND:3
	CASE 0
		LOCALS = 고간
	CASE 1
		LOCALS = 국부
	CASEELSE
		LOCALS = 가랑이
	ENDSELECT
ENDIF

RETURNF LOCALS


code:
;FileName_APPEND_SYS.ERB --------------------------- Rev1. 00
;era 명련사_시스템 변경계 이벤트 통괄
;CALL		USER
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
;EVENT_NO_LIST ============================== FLAG:1
;FLAG:1=1 사당 멤버 출현 이벤트
;FLAG:1=2 마미조 내방 이벤트
;FLAG:1=3 명련사멤버 추가 이벤트
;FLAG:1=4
;FLAG:1=5

@SYS_EVENT
;--- 이벤트 자기 정의 블록 ---

;--- 이벤트 실행 블록 ---

;	;자택에 SUMIKOMI_ROOM가 설정되어 있다
	IF FLAG:더부살이캐릭터== 0 && !FLAG:더부살이이벤트 && SUMIKOMI_ROOM && CFLAG:MASTER:현재위치 != OMANEKIBEYA()
		CALL NEWMEMBER_ADD
	ENDIF
;ENDIF

;이하, 각 이벤트 함수 ===============================

;더부살이 EX캐릭터 결정 이벤트
@NEWMEMBER_ADD
#DIM 인원수
#DIM ROOM_N
#DIM C_ID
VARSET LOCAL
인원수 = 0
FOR LOCAL, 1, CHARANUM
	;이미 살고 있는 멤버는 제외한다
	SIF CFLAG:LOCAL:신사거주 > 0
		CONTINUE
	IF TALENT:LOCAL:연모 && EXP:LOCAL:애정경험 >= 200
		인원수 ++
		LOCAL:LOCAL = 1
		LOCALS = %CALLNAME:LOCAL%
	ENDIF
NEXT
SIF !인원수
	RETURN
PRINTL
DRAWLINE
PRINTFORML 아침, 눈을 뜬 %CALLNAME:MASTER% 앞으로 내방자가 나타났다

PRINTFORML %LOCALS%\@ 인원수 > 1 ?  들은 # %조사만처리(LOCALS,"은")% \@ %CALLNAME:MASTER%에게 %GET_MAPNAME(MAIN_MAP)%에서 살고 싶다고 고했다
PRINTL
PRINTFORML %NAME_FROM_PLACE(SUMIKOMI_ROOM)%을 개장하면, 한사람 정도 살 수 있을 것이다
PRINTFORMW %CALLNAME:거점_주인%들은 제대로 설명하면 환영해 줄 것 같다
PRINTFORML %마스터는%…
C_ID = 1
FOR LOCAL, 1, CHARANUM
	;이미 살고 있는 멤버는 제외한다
	SIF CFLAG:LOCAL:신사거주 > 0
		CONTINUE
	IF TALENT:LOCAL:연모 && EXP:LOCAL:애정경험 >= 200
		C_ID = LOCAL
		PRINTFORML [{LOCAL}] - %CALLNAME:LOCAL%의 이사를 인정한다
	ENDIF
NEXT
PRINTFORML [999] - 인정하지 않는다
$INPUT_LOOP
INPUT
IF RESULT == 999
	PRINTFORMW %마스터가% 고개를 저으면 %LOCALS%\@ 인원수 > 1 ? 들은 # %조사만처리(LOCALS,"은")% \@ 유감스럽게 숙여 %CALLNAME:MASTER%의 방을 뒤로 했다…
	;이 방식이라면 한사람 밖에 말을 부를 수 없지만 좋은 방법은 없는 것인가
	;우선 다수의 경우에 각각의 캐릭터의 말을 호출할 수 있도록(듯이) 해 본다, 다만 실제로 할 수 있을지 어떨지는 테스트 되어 있지 않기 때문에 자신은 없다
	IF 인원수 > 1
		FOR LOCAL, 1, CHARANUM
			;이미 살고 있는 멤버는 제외한다
			SIF CFLAG:LOCAL:신사거주 > 0
				CONTINUE
			IF TALENT:LOCAL:연모 && EXP:LOCAL:애정경험 >= 200
				CALL KOJO_MESSAGE_SEND("EVENT", 19, LOCAL, 2,0)
				PRINTL
			ENDIF
		NEXT
	ELSE
		CALL KOJO_MESSAGE_SEND("EVENT", 19, C_ID, 2,0)
	ENDIF
	RETURN 0
ELSEIF RESULT >= CHARANUM
	CLEARLINE 1
	GOTO INPUT_LOOP
ELSEIF !LOCAL:RESULT
	CLEARLINE 1
	GOTO INPUT_LOOP
ENDIF
PRINTFORML %CALLNAME:RESULT%의 손을 잡아, %CALLNAME:거점_주인%에 부탁해 보는 일을 고하면 %조사처리(CALLNAME:RESULT,"는")% 기쁜듯이 미소짓고 있다
PRINTFORMW 그 다리로 %CALLNAME:거점_주인%의 원래로 간절히 부탁하러 가면 기분 좋게 승낙해 주었다…
PRINTL
PRINTFORMW - %조사처리(CALLNAME:RESULT,"가")% 새롭게 %GET_MAPNAME(MAIN_MAP)%의 동료로서 더해졌습니다 -
PRINTL
FLAG:더부살이캐릭터 = RESULT
CFLAG:(FLAG:더부살이캐릭터):311 = SUMIKOMI_ROOM
CFLAG:(FLAG:더부살이캐릭터):신사거주 = SUMIKOMI_ROOM
STR:16 = %CALLNAME:RESULT%의 오두막
CALL KOJO_MESSAGE_SEND("EVENT", 19, RESULT, 1,0)

CALL ROOMSETTING


code:
;FLAG:50:bit 표기:0마리사, 1, 2대, 3⑨,  4사쿠야, 5 레밀리아, 6 앨리스, 7 흰 백합, 8 패모속의 백합,9-10-11프리즘리바즈, 12요요우무13-14-15야쿠모가, 16 리글, 17―,  18아야야, 19에에이키 님20-21-22산신님 서, 23 텐시, 24이쿠 씨,25-26-27지령조, 28 나즈,  29단지들 -응, 30네, 31은 세우지 않아, 32카센, 33 엘렌, 34 리카코, 35메이라,
;FLAG:50:bit 표기:36리카, 37 루이즈, 38 사토리 님, 39 플랑 , 40니토리, 41레이센, 42, 43 파체, 44 팔꿈치 링, 45 미코 님, (46 코코로, 의미 없음), 47에링, 48개, 49 파르시, 50도 고담, 51 공주님, 52카게로, 53유기, 54화, 55님, 56-저기,  57할까 링, 58 마미조, 59 코스즈, (60신묘마루·의미 없음), 61에이린, 62세키반키, 63 레티
;FLAG:51:bit 표기:0메디,  1코마치, 2시즈하, 3미노리코, 4히나, 5아큐,  6렌코, 7 메리, 8 키스메,  9야마메, 10 이치린, 11 무라사, 12쇼,  13쿄코,  14요시카,  15냥냥, 16 토지, 17 후토, 18 와카사기, 19벤벤, 20 야츠하시, 21 라이코, 22 세이자, 23,24,25,26,27,28,29,30,31,32,33,34,35,
;FLAG:51:bit 표기:
;FileName_BEFORETRAIN.ERB -------------------------- Rev1. 00
;아침 1일 스타트 처리
;CALL		SYSTEM
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@EVENTTRAIN
#PRI
#DIM 콜렉션스키마입수수
#DIM 역서입수수
#DIM 수면시간
#DIM 당신의기상시간
#DIM 의뢰주후보
#DIM 의뢰인
#DIM 동거사람
#DIM 추가수면시간
#DIM 기상방해확률
#DIM 安眠補正 ;숙면 보정
#DIM 快眠係数 ;숙면 계수

;TRAGET 초기화
VARSET TARGET
VARSET LOCAL
;아침 1일 스타트 처리
;초기화

FLAG:일일이벤트 = RAND:1000
FLAG:70 = 0
RESETBGCOLOR
CFLAG:MASTER:동침중 = 0
IF IN_HOME(CFLAG:MASTER:초기위치) == 0 || CFLAG:MASTER:초기위치 <= 0
	PRINTW 초기위치가 부정합니다. Default의 위치에 되돌립니다.
	CFLAG:MASTER:초기위치 = Default초기위치
ENDIF
TFLAG:102 = 1
TFLAG:연회장 = GET_JOBPLACE("연회")
UFUFU_START_TIME = 0
UFUFU_START_DAY = 0
FOR LOCAL,0,CHARANUM
	CALL InitializeBeforeTrain(LOCAL)
NEXT

;EX캐릭터 방명재정의

SIF CFLAG:MASTER:초기위치 == 53
	STR:53 = %CALLNAME:MASTER%의 창고
SIF FLAG:16
	STR:55 = %CALLNAME:(FLAG:16)%의 방
FLAG:창고잠금 = 0

CALL RESET_DISH
CALL CHANGE_TIMEZONE
;시간이 MASTER의 기상 예정 시각보다 크면 일자＋1
IF TIME > TIME:3
	DAY ++
	DAY:3 ++
	TIME -= 1440
	CALL CHANGE_MONTH
	;연회의 참가자 결정
	SIF FLAG:3 && FLAG:연회개최플래그 != 98 && FLAG:연회개최플래그 != 99 && DAY >= FLAG:34
		CALL ENKAI_ENTRY
ENDIF

;아가씨의 몸 주기의 진행
;=================================
FOR LOCAL,0,CHARANUM
	;키스경험이 0이라면[키스미경험]을 취득(미설정캐릭터에게로의 대책이라고 생각된다)
	SIF EXP:LOCAL:키스경험 == 0 && TALENT:LOCAL:키스미경험 == 0
		 TALENT:LOCAL:키스미경험 = 1
	;여성기 부재(성별의 bit0가 0) 혹은[애같음][유치]로[초경]미취득 이라면 주기를 설정하지 않는다
	IF ((TALENT:LOCAL:150 || TALENT:LOCAL:151) && !TALENT:LOCAL:초경) || !GETBIT(TALENT:LOCAL:성별, 0)
		CFLAG:LOCAL:900 = 0
		CONTINUE
	ENDIF
	;주기가 미설정의 경우 여기서 결정
	SIF (CFLAG:LOCAL:900 == 0 || CFLAG:LOCAL:900 > 11)
		CFLAG:LOCAL:900 = RAND:10
	;주기가  1진행된다
	CFLAG:LOCAL:900 += 1
	;아직 10 일주기(11이상으로 1의 위인 만큼 한다)
	SIF CFLAG:LOCAL:900 > 10
		 CFLAG:LOCAL:900 %= 10
	SIF TALENT:LOCAL:임신 > 0
	 	TALENT:LOCAL:임신 += 1
NEXT
;=================================

CALL 내방플래그

;EX캐릭터 휴일 데이트처 우연히 판정
CALL 만남판정

동거사람 = 0
추가수면시간 = 0
;이불에 질질 끌어 들여진다
IF CFLAG:MASTER:현재위치 == OMANEKIBEYA()
	동거사람 = CFLAG:MASTER:초대
ELSEIF PRIVATEROOM:(CFLAG:MASTER:초기위치 % 100) && CFLAG:MASTER:현재위치 == CFLAG:MASTER:초기위치
	동거사람 = PRIVATEROOM:(CFLAG:MASTER:초기위치 % 100)
ENDIF

IF TALENT:동거사람:연모 && INRANGE(TIME:3, CFLAG:동거사람:기상시간 - 180, CFLAG:동거사람:기상시간)
	기상방해확률 = 20
	SIF DAY:2 == 2
		기상방해확률 = 10
	SIF DAY:2 == 4
		기상방해확률 = 35
	IF 기상방해확률 > RAND:100
		PRINTFORML %마스터가% 엉금엉금 이불에서 기어 나오려고 하자, %CALLNAME:동거사람%에 껴안겨졌다
		PRINTFORML 조금 더 함께 자고 싶은 것 같다
		PRINTFORMW 어쩔 수 없기 때문에 그녀가 일어날 때까지 이불에서 보내기로 했다……
		EXP:동거사람:애정경험 += 5
		CALL COLORMESSAGE(@"애정경험＋5 (%CALLNAME:동거사람%)", C_PINK, 1)
		추가수면시간 = CFLAG:동거사람:기상시간 - TIME:3
	ENDIF
ENDIF

;30분 마다 신사의 캐릭터를 이동시킨다
LOCAL = 1
WHILE LOCAL
	IF TIME + 30 < TIME:3 + 추가수면시간
		TIME:1 = TIME + 1440 * DAY
		TIME += 30
		CALL CHARA_MOVEMENT
	ELSE
		;전캐릭터의 현재위치 보존
		FOR LOCAL:1,0,CHARANUM
			CFLAG:(LOCAL:1):전턴위치 = CFLAG:(LOCAL:1):현재위치
		NEXT
		;시간을 MASTER의 기상 예정 시각에
		TIME =  TIME:3 + 추가수면시간
		TIME:1 = TIME + 1440 * DAY
		TFLAG:당신기상 = TIME + 1440 * DAY
		CALL CHARA_MOVEMENT
		BREAK
	ENDIF
WEND

;눈을 뜨고 공통 메시지 표시
CALL GOODMORNING

FOR LOCAL,1,CHARANUM
	;초기위치는 99로 통일(53은 낡은 데이터용)
	SIF CFLAG:LOCAL:초기위치 == 53 || CFLAG:LOCAL:초기위치 == MAXROOM || CFLAG:LOCAL:초기위치 == 0
		CFLAG:LOCAL:초기위치 = SUKIMA()
	SIF CFLAG:LOCAL:신사거주
		CFLAG:LOCAL:초기위치 = CFLAG:LOCAL:신사거주
	;일을 정의
	BASE:LOCAL:일량 = MAXBASE:LOCAL:일량
	;불러 하고 있으면(자) 만남 플래그를 자택위치로 한다
	IF CFLAG:LOCAL:초대
		[SKIPSTART]
		;연동 캐릭터는 워프해 버리므로 일괄 보정
		;↑이것 어떤 의미이겠지, 구문 보는 한 마지막 FLAG:(590 + LOCAL) = CFLAG:LOCAL:자택위치만으로 좋은 생각이 드는데
		SELECTCASE LOCAL
		;프리즘리버
		CASE 20, 21, 22
			FLAG:(590 + 20) = CFLAG:LOCAL:자택위치
			FLAG:(590 + 21) = CFLAG:LOCAL:자택위치
			FLAG:(590 + 22) = CFLAG:LOCAL:자택위치
		;아키 자매
		CASE 77, 78
			FLAG:(590 + 77) = CFLAG:LOCAL:자택위치
			FLAG:(590 + 78) = CFLAG:LOCAL:자택위치
		;연꽃 메리
		CASE 81, 82
			FLAG:(590 + 81) = CFLAG:LOCAL:자택위치
			FLAG:(590 + 82) = CFLAG:LOCAL:자택위치
		;츠쿠모 자매(자택은 없지만 설정했을 때 용)
		CASE 94, 95
			FLAG:(590 + 94) = CFLAG:LOCAL:자택위치
			FLAG:(590 + 95) = CFLAG:LOCAL:자택위치
		CASEELSE
			FLAG:(590 + LOCAL) = CFLAG:LOCAL:자택위치
		ENDSELECT
		[SKIPEND]
		CFLAG:LOCAL:만난위치 = CFLAG:LOCAL:자택위치
		;MASTER는 외박 하고 있으면(자) 데이트 플래그(외출 플래그)를 세운다
		SIF CFLAG:MASTER:초대
			CFLAG:MASTER:데이트중 = CFLAG:LOCAL:자택위치 / 100
		;그 날은 연속해서 묵을 수 없게
		TCVAR:LOCAL:연속숙박금지 = 1
	ENDIF
NEXT

;시스템 변화계 이벤트
CALL SYS_EVENT

;기상시간의 기록
당신의기상시간 = 1440 * DAY + TIME

;체력 기력의 회복
IF DAY == 1
	수면시간 = 480
ELSE
	수면시간 = 당신의기상시간 - CFLAG:MASTER:취침시간 - 추가수면시간 / 2
ENDIF
PRINTL
IF FLAG:기절중단
	SIF CFLAG:MASTER:현재위치 != ENTRANCE
		CFLAG:MASTER:데이트중 = FLAG:기절중단 - 100
	FLAG:기절중단 = 0
	수면시간 /= 5
	PRINTFORMW 이상한 장소에서 잤더니 등이 아프다……
ELSEIF CFLAG:MASTER:철야
	수면시간 /= 2
	PRINTFORMW 어젯밤은 잘 못잤다……
ELSE
	PRINTFORMW {수면시간 / 60}시간 잤다
ENDIF

CALL SleepSoundly
安眠補正 = RESULT
快眠係数 = 480 - (60 * TALENT:MASTER:회복속도) - 安眠補正
BASE:MASTER:체력 += MAXBASE:MASTER:체력 * 수면시간 / 2 / 快眠係数
BASE:MASTER:기력 += MAXBASE:MASTER:기력 * 수면시간 / 快眠係数
BASE:MASTER:정력 += MAXBASE:MASTER:정력 * 수면시간 / 快眠係数

BASE:MASTER:체력 = MIN(MAXBASE:MASTER:체력, BASE:MASTER:체력)
BASE:MASTER:기력 = MIN(MAXBASE:MASTER:기력, BASE:MASTER:기력)
BASE:MASTER:정력 = MIN(MAXBASE:MASTER:정력, BASE:MASTER:정력)
BASE:MASTER:발기 = MIN(1000 * BASE:MASTER:체력 / MAXBASE:MASTER:체력, 1000)
;DEBUGPRINTFORM 기력{MAXBASE:MASTER:기력 * 수면시간 / 2 / (480 -120 * TALENT:MASTER:회복속도)}회복
CFLAG:MASTER:철야 = 0

;의뢰의 발생과 소멸
CALL IRAI_HAPPEN
CALL IRAI_VANISH
CALL MEMO_IRAI_DEADLINE

IF FLAG:운세뽑기계약
	DRAWLINE
	PRINTFORML
	PRINTFORML ┌────────┐
	PRINTFORML ｜　　　　　　　　｜
	PRINTFORML ｜　　멋　진　　　｜
	PRINTFORML ｜　　운세뽑기　　｜
	PRINTFORML ｜　　　　　　　　｜
	PRINTFORML └────────┘
	PRINTFORMW
	CALL FORTUNE
	PRINTFORML
	DRAWLINE
ENDIF
IF FLAG:콜렉션스키마 != NOW_PERIOD() ;이하 NOW_PERIOD() 부분은 (DAY:0)/5 에서 수정됨
	IF FLAG:콜렉션스키마 < NOW_PERIOD()
		콜렉션스키마입수수 = 2
		역서입수수 = 1
	ELSE
	콜렉션스키마입수수 = (FLAG:콜렉션스키마 - NOW_PERIOD()) * 2
	역서입수수 = FLAG:콜렉션스키마 - NOW_PERIOD()
	ENDIF
	ITEM:콜렉션의스키마 = MIN(콜렉션스키마입수수 + ITEM:콜렉션의스키마, 10)
	ITEM:역서 = MIN(역서입수수 + ITEM:역서, 5)
	FLAG:콜렉션스키마 = NOW_PERIOD()
	PRINTL
	PRINTFORML ☆콜렉션의스키마가 조금 보충되어 있었다(현재：{ITEM:콜렉션의스키마}개)
	PRINTFORML ☆역서가 조금 보충되고 있었다(현재：{ITEM:역서}개)
ENDIF
IF !ITEM:미라다의돌 && (FLAG:미라다지급 < NOW_PERIOD()) ;KR판 구잔재. 일판에서는 주석처리
	CALL COLORMESSAGE(@"애욕을 독점하는 미라다의돌밀레니엄은 알고 있어? ", C_D_PINK, 2)
	PRINTFORMW ☆애욕을 독점하는 미라다의돌 밀레니엄이 지급되었다
	ITEM:미라다의돌 ++
	FLAG:미라다지급 = NOW_PERIOD()
ENDIF
;소작인 또한 월초(첫날 제외하다)라면 연공이 내세우고
;외박 해 도망 다니는 일도 할 수 없지는 않다
SIF FLAG:지주 && FLAG:연공 && (TARGET == FLAG:지주 || CFLAG:MASTER:현재위치 != OMANEKIBEYA())
	CALL NENGU_NOUMIN
CALL KODUCHI
CALL CHARA_MOVEMENT

;기상시에 머리맡에의 선물 판정을 실시한다
;CALL GET_GIFT

;조교자는 누군가
IF ASSIPLAY == 0
	PLAYER = MASTER
ELSE
	PLAYER = ASSI
ENDIF
;--------------------------------------------
;대청소
FOR LOCAL,1,11
SIF CFLAG:LOCAL:335 > 0
	CFLAG:LOCAL:335 -= 1
NEXT
SIF CFLAG:MASTER:335 > 0
	CFLAG:MASTER:335 -= 1

FOR LOCAL, 0, CHARANUM
	;의상 설정
	;행동중에 입는 옷을 설정해 갈아 입을 옷 한다
	IF LOCAL == MASTER || !수면시간(LOCAL)
		CALL SELECT_CLOTHES(LOCAL, "오늘의 의상")
		CALL SELECT_CLOTHES(LOCAL, "오늘의 파자마")
		CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_평상복")
		SETBIT CFLAG:LOCAL:갈아입음실행, 0
	ELSE
		CALL SELECT_CLOTHES(LOCAL, "내일의 의상")
		CALL SELECT_CLOTHES(LOCAL, "오늘의 파자마")
		CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_파자마")
		CFLAG:LOCAL:파자마 = 1
		SETBIT CFLAG:LOCAL:갈아입음실행, 1
	ENDIF
	IF SPECIAL_OUTFIT(LOCAL)
		CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_의상세트", SPECIAL_OUTFIT(LOCAL))
		[IF_DEBUG]
		PRINTFORML %CALLNAME:LOCAL%는 평상시와 다른 옷으로 갈아입었습니다
		[ENDIF]
		TCVAR:LOCAL:어나더_평상복 = 1
	ENDIF
	CALL 속옷확인리셋(LOCAL)
	CALL CLOTHES_SETTING_TRAIN(LOCAL)
NEXT

;발원함 리셋
CVARSET CFLAG, GETNUM(CFLAG, "다음날내방"), 0
FLAG:560 = 0
FLAG:561 = 0
IF FLAG:일시체재
	CFLAG:(FLAG:일시체재):초기위치 = CFLAG:(FLAG:일시체재):본래의초기위치
	CFLAG:(FLAG:일시체재):신사거주 = 0
	CFLAG:(FLAG:일시체재):거점내방 = 1
	FLAG:일시체재 = 0
ENDIF
;----------------------------------------------

;----------------------------------------------
;FileName_BEFORETRAIN.ERB -------------------------- Rev1. 00
;눈을 뜨고 공통 메시지 표시
;CALL		USER
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@GOODMORNING
#DIM 기상상황
#DIM 기상장소
#DIM 연회위치시정
#DIM  DYNAMIC CNT_BIRTHDAY
#DIMS DYNAMIC NAME_CHILD   , 10 * 캐릭터수상한 + 1
#DIM  DYNAMIC AGE_CHILD    , 10 * 캐릭터수상한 + 1
VARSET LOCAL
TARGET = 0
기상상황 = 0
기상장소 = 0

;MASTER와 동실의 캐릭터를 찾는다
FOR LOCAL,1,CHARANUM
	SIF CFLAG:MASTER:현재위치 == CFLAG:LOCAL:현재위치
		;TARGET 결정
		TARGET = LOCAL
NEXT
;타겟이 들어가면 표시
IF TARGET
	;IF CFLAG:TARGET:수면
	IF 수면시간(TARGET)
		;동거중
		IF CFLAG:MASTER:초기위치 == CFLAG:TARGET:초기위치
			;동거중 또한 상대가 자고 있다
			기상상황 = 3
			FOR LOCAL,0,CHARANUM
				CALL SET_TOGETHER(0,LOCAL,1)
			NEXT
			;MASTER와의 접촉 위치를 리셋
			CALL TOUCH_SET(0,0,0,1)
			;RETURN 1
		;비동거 중(방에서 나가므로 기상구상은 없음) 초대 취침 조건인 연모에게 왜 나가 버리는지 모르기 때문에 잠입이나 장난 때만 철퇴하도록 변경했습니다.		ELSE
		;덧붙여서 말하면 먼저 방주인이 밖에 나와 있을 경우, 방에 남겨진다. 만약 수면간 후 취침을 실장했을 때에는 여기서 요분기
		ELSE
			FOR LOCAL,0,CHARANUM
				CALL SET_TOGETHER(0,LOCAL,1)
			NEXT
			;MASTER와의 접촉 위치를 리셋
			CALL TOUCH_SET(0,0,0,1)
			;내 방이 아닌, 억울할 경우 이동 가능한 곳으로 이동
			SIF (CFLAG:TARGET:은밀중 || CFLAG:TARGET:장난) && CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
			CALL GETOUT(MASTER)
			;잠입해 OFF
			CALL SET_SNEAK(0,TARGET,2)
			;장난 OFF
			CALL SET_PRANK(0,TARGET,2)
		ENDIF
	ELSE
		;동거중
		IF CFLAG:MASTER:초기위치 == CFLAG:TARGET:초기위치
			;동거중 또한 상대가 일어나고 있다
			기상상황 = 2
			FOR LOCAL,0,CHARANUM
				CALL SET_TOGETHER(0,LOCAL,1)
			NEXT
			;MASTER와의 접촉 위치를 리셋
			CALL TOUCH_SET(0,0,0,1)
		;비동거중
		ELSE
			;비동거(同棲) 한편 상대가 일어나고 있다
			기상상황 = 1
			;MASTAR의 자기 방에서 기상했다
			IF CFLAG:MASTER:현재위치 == CFLAG:MASTER:311
				기상장소 = 1
				FOR LOCAL,0,CHARANUM
					CALL SET_TOGETHER(0,LOCAL,1)
				NEXT
				;MASTER와의 접촉 위치를 리셋
				CALL TOUCH_SET(0,0,0,1)
			;TARGET의 자기 방에서 기상했다(연모 첨부 상대의 방에서 동침 후 취침시만의 괄
			ELSEIF CFLAG:MASTER:현재위치 == CFLAG:TARGET:311
				기상장소 = 0
				FOR LOCAL,0,CHARANUM
					CALL SET_TOGETHER(0,LOCAL,1)
				NEXT
				;MASTER와의 접촉 위치를 리셋
				CALL TOUCH_SET(0,0,0,1)
			;그 이외(
			ELSE
				;어디서 자...?연모있는 캐릭터의 방에 다른 연모있는 캐릭터를 데리고 잤다..???
				기상장소 = -1
			ENDIF
			;잠입해 OFF
			CALL SET_SNEAK(0,TARGET,2)
			;장난 OFF
			CALL SET_PRANK(0,TARGET,2)
		ENDIF
		;CALL KOJO_EVENT(6, TARGET, 1)
		;동침 후인가 어떤가
		;동침 플래그가 서지 않기 때문에, 통상 기상만
		;IF CFLAG:MASTER:동침중
			;CALL KOJO_MESSAGE_SEND("EVENT", 2, TARGET, 2,0)
		;ELSE
		;ENDIF
	ENDIF
	SIF 기상상황 == 0 || 기상장소 == -1
		CALL MESSAGE_GOODMONING(기상상황, 기상장소)
	SIF 기상상황 && 기상장소 != -1
		CALL KOJO_MESSAGE_SEND("EVENT",2,TARGET, 기상상황, 기상장소)
ENDIF
IF FLAG:이상기상
	PRINTFORML 대기에는 심상치 않은 기색이 감돌고 있다
ELSE
	SELECTCASE TIME:5
		CASE 0,1
			PRINTL 좋은 날씨다.
		CASE IS & 4
			PRINTL 비가 내리고 있다.
		CASE IS & 8
			PRINTL 눈이 내리고 있다.
	ENDSELECT
ENDIF
IF FESTIVAL() != ""
	PRINTFORML 오늘은 %조사처리(FESTIVAL(),"다")%.
	CALL SEASONAL_MORNING
ENDIF
IF FLAG:개시일 - DAY <= 0 && FLAG:연회개최플래그
;잔치날
	PRINTFORM 오늘은
	CALL 연회예정표시
	PRINTL 가 있다
	;여기서의 연회장 대입이라면 CHARA_MOVEMENT로 불안정이 나왔기 때문에 상부에 이동
	;FLAG:32 = GET_JOBPLACE("연회")
	IF TIME:3 >= FLAG:33
		FOR 연회위치시정, 1, CHARANUM
			IF CFLAG:연회위치시정:연회참가
				CFLAG:연회위치시정:직장 = TFLAG:연회장
				CFLAG:연회위치시정:현재위치 = TFLAG:연회장
			ENDIF
		NEXT
	ENDIF
ENDIF
IF CHILD_HAPPY_BIRTHDAY(CNT_BIRTHDAY, NAME_CHILD, AGE_CHILD)
	PRINTFORM 오늘은
	FOR LOCAL, 1, CNT_BIRTHDAY + 1
		PRINTFORM %NAME_CHILD:LOCAL%(만{AGE_CHILD:LOCAL}세)
		SIF LOCAL == CNT_BIRTHDAY
			BREAK
		PRINTFORM  %조사만처리(NAME_CHILD:LOCAL,"와")%
	NEXT
	PRINTFORML  의 생일이다
ENDIF
CALL CHARA_MOVEMENT
SIF CFLAG:MASTER:지금바쁨
	PRINTFORML …아무래도 자위를 한 채로 자 버린 것 같다…

CALL TODAYS_CHARISMA

;----------------------------------------------
@KODUCHI
;요술망치의 대상
IF CFLAG:MASTER:요술망치 > 0
	CFLAG:MASTER:마력회수 -= 1
	LOCAL = CFLAG:MASTER:마력회수
	IF CFLAG:MASTER:요술망치 == 8
			PRINTFORMW 신묘마루의 몸은 원래대로 돌아갔다
		TALENT:71:체형 = -5
		IF CFLAG:71:초기위치 == 54 || CFLAG:71:현재위치 == 9
			CFLAG:71:현재위치 = 54
			CFLAG:71:초기위치 = 54
		ENDIF
	ELSEIF CFLAG:MASTER:요술망치 == 4
		IF CFLAG:MASTER:마력회수 < 1
			TALENT:MASTER:형상 = CFLAG:MASTER:382
		ELSE
			PRINTW 그렇게 컸던 물건이 비참한 것이 되고 있다…
			TALENT:MASTER:형상 = 2
		ENDIF
	ELSEIF CFLAG:MASTER:요술망치 == 5
		IF CFLAG:MASTER:마력회수 < 1
			TALENT:MASTER:형상 = CFLAG:MASTER:382
		ELSE
			PRINTW 2개 있던 장대가……한 개에…어디 가 아니게 되고 있다
			TALENT:MASTER:183 = 0
			TALENT:MASTER:형상 = 2
			BASE:MASTER:정력 -= MAXBASE:MASTER:정력 / 2
		ENDIF
	ELSEIF CFLAG:MASTER:요술망치 == 6
		SIF CFLAG:MASTER:마력회수 == 2
			PRINTFORMW %CALLNAME:MASTER%의 몸은 원래대로 돌아가고 있다
		TALENT:MASTER:100 = FLAG:원래체격
		IF CFLAG:MASTER:초기위치 == 54 || CFLAG:MASTER:현재위치 == 54
			CFLAG:MASTER:현재위치 = 9
			SIF CFLAG:MASTER:초기위치 == 54
				CFLAG:MASTER:초기위치 = Default초기위치
		ENDIF
	ELSEIF CFLAG:MASTER:요술망치 == 7
		;원래 크기에 되돌릴 뿐(만큼)이므로 변화 없음
	ENDIF
	IF CFLAG:MASTER:마력회수 <= 0
		CFLAG:MASTER:요술망치 = 0
		CFLAG:MASTER:마력회수 = 0
	ENDIF
ENDIF

;----------------------------------------------
;아침인사 공통구상
;기상상황 = 0: 비동거상대 잠 1: 비동거상대일어남 2:동거상대일어남 3:동거중 상대잠 4:
;기상장소 = 0:상대 방 1:내 방
;----------------------------------------------
@MESSAGE_GOODMONING(기상상황,기상장소)
#DIM 기상상황
#DIM 기상장소
PRINTFORML
SELECTCASE 기상상황
CASE 0
	;非同棲相手眠り
	IF !기상장소
		;とにかくやましい時のみ抜け出すよう変更
		PRINTFORMW %마스터가% 눈을 뜨니 옆에서 %타겟이% 자고 있었다
		IF CFLAG:TARGET:은밀중 || CFLAG:TARGET:장난
			IF CFLAG:MASTER:현재위치 == CFLAG:MASTER:초기위치
				PRINTFORMW %마스터는% %타겟이% 일어났을 때 어떻게 할지 생각했다…
			ELSE
				PRINTFORMW %타겟이% 눈치채지 못하게 %마스터는% 천천히 방을 빠져 나갔다…
			ENDIF
		ELSE
			PRINTFORMW %마스터는% 조용하게 %CALLNAME:TARGET%의 옆을 빠져 나가 하루의 준비를 시작했다
		ENDIF
	ELSE
		;今のところ死に分岐
		PRINTFORMW %마스터가% 눈을 뜨자 옆에서 %타겟이% 자고 있었다
		PRINTFORMW 어째서 %타겟이% 방에 있는 것일까 라고 조금 놀라면서도\@ BEDROOM(CFLAG:MASTER:현재위치) ? 이불 # 옆 \@을 빠져 나가, 하루의 준비를 시작했다
	ENDIF
CASE 1
	;非同棲相手起き
	IF 기상장소 == 0
		;초대방수면かな…？ 연모付き確定？
		PRINTFORMW %마스터가% 눈을 뜨자 %타겟과% 눈이 마주쳤다
		PRINTFORMW 아침의 인사를 한 후, %CALLNAME:TARGET%의 근처를 빠져 나가 하루의 준비를 시작했다
	ELSEIF 기상장소 == 1
		;自室にキャラを連れ込んだ後で（あなたの수면時間の方が早い）寝た場合や勝手に部屋に入ってくる場合（三月精とか割と頻繁）
		;自分で連れ込んだんじゃろがいっ！ って時は個人口上側で対処してください
		PRINTFORMW %마스터가% 눈을 뜨자 %타겟과% 눈이 마주쳤다
		PRINTFORMW 어째서 %타겟이% 방에 있는 것일까 라고 조금 놀라면서도\@ BEDROOM(CFLAG:MASTER:현재위치) ? 이불 # 옆 \@을 빠져 나가, 하루의 준비를 시작했다
	ELSE
		;ここはどこ、この子は誰状態 どうやったらこの分岐に来るかわからない…
		PRINTFORMW 어째서 이런 곳에서 자고 있고, 어째서 %타겟이% 옆에 있는 것일까 하는 의문이 들면서도,\@ BEDROOM(CFLAG:MASTER:현재위치) ? 이불 # 옆 \@을 빠져 나가 하루의 준비를 시작했다
	ENDIF
CASE 2
	;同棲相手起き
	PRINTFORMW %마스터가% 눈을 뜨자  %타겟과% 눈이 마주쳤다
	PRINTFORMW 잠시 동안 %타겟과% 아침의 인사를 주고 받고 나서\@ BEDROOM(CFLAG:MASTER:현재위치) ? 이불 # 옆 \@을 빠져 나가 하루의 준비를 시작했다
CASE 3
	;同棲相手眠り
	PRINTFORMW %마스터가% 눈을 뜨자 옆에서 아직 %타겟이% 새근새근 자고 있었다
	PRINTFORMW %마스터가% %타겟을% 깨우지 않도록 \@ BEDROOM(CFLAG:MASTER:현재위치) ? 이불 # 옆 \@을 빠져 나가 하루의 준비를 시작했다
ENDSELECT
PRINTFORML
RETURN 1

;----------------------------------------------

@InitializeBeforeTrain(ARG)
CFLAG:ARG:면간발각 = 0
CFLAG:ARG:전턴위치 = 0
CFLAG:ARG:데이트중 = MAIN_MAP
CFLAG:ARG:졸음쫒기강도 = 0
;사정、발기、무드、이성、분노
BASE:ARG:사정 = 0
BASE:ARG:발기 = 0
BASE:ARG:무드 = 0
BASE:ARG:이성 = MAXBASE:ARG:11
BASE:ARG:분노 = 0
BASE:ARG:술기운 = 0
BASE:ARG:발기 = 0
BASE:ARG:심도 = 0
CFLAG:ARG:발기도 = 0
;행동はリセット
CFLAG:ARG:행동 = 0
;덮침侵入リセット
CLEARBIT CFLAG:ARG:42,8
;경험표시用
FOR LOCAL:1,0,100
	TCVAR:ARG:(400 + LOCAL:1) = EXP:ARG:(LOCAL:1)
NEXT
CALL SETTING_PERSONALITY(ARG)
;性交フラグ
TEQUIP:ARG:50 = -1
TEQUIP:ARG:51 = -1
TCVAR:ARG:100 = -1
TCVAR:ARG:101 = -1
;中出し表示用フラグ
TCVAR:ARG:103 = -1
TCVAR:ARG:105 = -1
;수면제차関連
TCVAR:TARGET:수면제차사용 = 0
;사정処理用文字列変数
CSTR:ARG:10 = /
CSTR:ARG:11 = /
IF LOCAL && CFLAG:ARG:현재위치 != OMANEKIBEYA()
	SIF ((CFLAG:MASTER:취침시간 - DAY * 1440) && (CFLAG:MASTER:취침시간 - DAY * 1440) < CFLAG:ARG:취침시간)
		CFLAG:ARG:현재위치 = CFLAG:ARG:초기위치
ENDIF
SIF IN_HOME(CFLAG:ARG:초기위치) != 1 & CFLAG:ARG:수면
	CALL SET_SLEEP(0,ARG,0)
RESET_STAIN ARG

@SleepSoundly
#DIM 安眠補正
SIF FLAG:기절중단
	RETURN 0
SIF CFLAG:MASTER:철야
	RETURN 0
SIF CFLAG:MASTER:초대
	RETURN 0
;安眠枕の補正
IF ITEM:스위티숙면배개
	安眠補正 = MIN(RAND:(SQRT(FLAG:튜토리얼글누계 + 1)), 40)
	SELECTCASE 安眠補正
		CASE IS > 30
			PRINTFORMW 배개 덕분인지 굉장한 느낌의 기상이다!
		CASE IS > 20
			PRINTFORMW 배개 덕분인지 피로가 꽤나 풀린 듯한 느낌이다.
	ENDSELECT
ELSE
	安眠補正 = 0
ENDIF
;ベッドの補正
IF ITEM:오동나무침대
	SELECTCASE ITEM:오동나무침대
		CASE 1
			安眠補正 += RAND:20 + RAND:20
		CASE 2
			安眠補正 += RAND:30 + RAND:30
		CASE 3
			安眠補正 += RAND:40 + RAND:40
	ENDSELECT
ENDIF

RETURN 安眠補正


code:
;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트 사키나 취득 함수
;CALL		USER
;ARG		VOID
;RETURN		데이트 사키나
;COMMENT
;-----------------------------------------------------------
@DATENAME_PLACE()
#FUNCTIONS
SIF TFLAG:망상중 == 1
	RETURNF "망상"
SIF CHK_DATENOW(CFLAG:MASTER:데이트중) == 0
	RETURNF ""
RETURNF STR:(CFLAG:MASTER:데이트중 + 5000)

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트 지점 나토리 이득 함수
;CALL		USER
;ARG		VOID
;RETURN		데이트 지점명
;COMMENT
;-----------------------------------------------------------
@DATENAME_SPOT()
#FUNCTIONS
SIF CHK_DATENOW(CFLAG:MASTER:데이트중) == 0
	RETURNF ""
SIF CFLAG:MASTER:현재위치 % 100 / 10 == 0
	RETURNF "도중"
RETURNF NAME_FROM_PLACE(CFLAG:MASTER:현재위치 / 10 + 6000)

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트 지점 번호 취득 함수
;CALL		USER
;ARG		VOID
;RETURN		데이트 지점 NO
;COMMENT
;-----------------------------------------------------------
@DATE_SPOT()
#FUNCTION
RETURNF CFLAG:MASTER:현재위치 % 100 / 10

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트처 도착 남은 시간 취득 함수
;CALL		USER
;ARG		VOID
;RETURN		데이트 지점 NO
;COMMENT
;-----------------------------------------------------------
@DATE_REMAINTIME()
#FUNCTION
RETURNF ((CFLAG:MASTER:현재위치 % 100) % 10) * 10

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트처 아르바이트 가부
;CALL		USER
;ARG		현재위치
;RETURN		아르바이트처 종류
;COMMENT
;-----------------------------------------------------------
@DATE_PARTTIME(ARG)
#FUNCTION
SELECTCASE ARG
	;특수 아르바이트 앞
	CASE 240,450,520,540,650,830
		RETURNF ARG
	;%데이트처%로 아르바이트 해∼…
	CASE 220,230,320,330,460
		RETURNF 2
	;%데이트처%로 식재료 모음을 해∼…
	CASE 350,410,430,440,510,530,630,710,810
		RETURNF 1
ENDSELECT
RETURNF 0

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트처 모브 존재 판정
;CALL		USER
;ARG		현재위치
;RETURN		통로가 많으면 1
;COMMENT	MASTER의 눈에 보이는 범위에 모브캐라(소녀 한정)가 걷고 있는(일 것이다) 장소인가 아닌가
;-----------------------------------------------------------
@DATE_HITOGOMI(ARG)
#FUNCTION
SIF !ARG
	ARG = CFLAG:PLAYER:현재위치
IF !AT_HOME(MASTER)
	SELECTCASE ARG
		;모브가 있다(와 마음대로 상상하고 있는 장소
		;지저(큰길, 온천, 지령전) 요괴의 산 산정(모리야신사본전, 경내, 구천의폭포) 요괴의 산 산록(없음) 삼도천~명계(중유의길) 마법의숲(없음) 미혹의 죽림(영원정) 홍마관(객실, 식당, 대도서관) 인간 마을(혼숙여관 이외) 명련사(지장길, 수양버들아래, 명련사경내, 명련사본당)
		CASE 960,940,930,850,840,810,610,450,370,360,350,270,250,240,230,220,210,140,130,120,110
			RETURNF 1
	ENDSELECT
ENDIF
RETURNF 0

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 00
;데이트처 모브 종류 존재 판정
;CALL		USER
;ARG		현재위치
;RETURN
;COMMENT	MASTER의 눈에 보이는 범위에 모브캐라(소녀 한정)가 걷고 있는(일 것이다) 장소인가 아닌가, 남자는 Gun 무시, 홉고블린은 신사이므로 소녀가 모여 꺅 캬 하고 있는 곳에는 접근하지 않습니다
;-----------------------------------------------------------
@DATE_HITOGOMI_SYUBETU(ARG)
#FUNCTIONS
VARSET LOCALS
SIF !ARG
	ARG = CFLAG:PLAYER:현재위치
SELECTCASE GET_MAPID(ARG)
	;모브의 종류
	;지저(큰길, 온천, 지령전)(큰길과 온천에서는 오니도 나온다
	CASE 9
		IF ARG != 960 && !RAND:3
			LOCALS += "오니"
		ELSE
			LOCALS += "요괴"
		ENDIF
	;요괴의 산 산정(모리야신사본전, 경내, 구천의폭포)(모리야신사경내만 드물게 인간
	CASE 8
		IF !RAND:5 && ARG == 840
			LOCALS += "인간"
		ELSE
			LOCALS += "요괴"
		ENDIF
	;요괴의 산 산록(없음)
	;삼도천~명계(중유의길)(사신, 요괴, 드물게 인간
	CASE 6
		IF !RAND:5
			LOCALS += "인간"
		ELSEIF !RAND:3
			LOCALS += "요괴"
		ELSE
			LOCALS += "사신"
		ENDIF
	;마법의숲(없음)
	;미혹의 죽림(영원정)(요괴 토끼, 드물게 인간(환자)
	CASE 4
		IF RAND:5
			LOCALS += "요괴토끼"
		ELSE
			LOCALS += "인간"
		ENDIF
	;홍마관(객실, 식당, 대도서관)(객실 식당은 요정 메이드, 대도서관은 소악마로 하고 싶지만, 개별 캐릭터화하고 있기 때문에 요정 메이드
	CASE 3
		LOCALS +=  "요정 메이드"
	;인간 마을(혼숙여관 이외)(기본 인간, 드물게 요괴
	CASE 2
		IF RAND:5
			LOCALS += "인간"
		ELSE
			LOCALS += "요괴"
		ENDIF
	;명련사(지장길, 수양버들아래, 명련사경내, 명련사본당)(인간과 요괴
	CASE 1
		IF !RAND:2
			LOCALS += "인간"
		ELSE
			LOCALS += "요괴"
		ENDIF
ENDSELECT
LOCALS += "의 소녀"

RETURNF LOCALS

;FileName_DATE_CMN.ERB ------------------------------- Rev1. 00
;데이트의 도중인가 어떤가
;CALL		USER
;ARG		현재위치
;-----------------------------------------------------------
@HALFWAY()
#FUNCTION
SIF CHK_DATENOW(CFLAG:MASTER:데이트중) && GET_DATE_PLACE(CFLAG:MASTER:현재위치) && CFLAG:MASTER:데이트중 != MAIN_MAP + 1 && CFLAG:MASTER:현재위치 != OMANEKIBEYA()
	RETURNF 1
RETURNF 0

;FileName_DATE_CMN.ERB ----------------------------- Rev1. 01
;데이트 돌아가 이벤트 (CFLAG:13)
;CALL		USER
;ARG		캐릭터 NO
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@DATE_EVENT(ARG)
#DIM 주부족
#DIM 경험부족

VARSET LOCAL
;캔슬 플래그(외부 참조에서 사용예:LOCAL@DATE_EVENT:99 = 1
LOCAL:99 = 0
;파라미터에 의한 발생 기초치 생성

LOCAL = MARK:ARG:괘씸각인 * 10 + MARK:ARG:쾌락각인 * 10
IF TALENT:ARG:연모
	LOCAL += MIN(ABL:ARG:친밀 * 10, 200) + MIN(ABL:ARG:욕망 * 10,200) + MIN(ABL:ARG:봉사정신 * 10,200)
	LOCAL += MIN(EXP:ARG:데이트경험, 200) + MIN(EXP:ARG:애정경험, 200)
ELSEIF TALENT:ARG:사모
	LOCAL += MIN(ABL:ARG:친밀 * 10, 100) + MIN(ABL:ARG:욕망 * 10,100) + MIN(ABL:ARG:봉사정신 * 10,100)
	LOCAL += MIN(EXP:ARG:데이트경험, 100)
ENDIF
LOCAL += GETPALAMLV(PALAM:ARG:욕정, 10) * 5 + GETPALAMLV(PALAM:ARG:호의, 10) * 5 + GETPALAMLV(PALAM:ARG:복종, 10) * 5
;60
LOCAL += BASE:ARG:무드 / 50 + (1000 - BASE:ARG:이성) / 30
LOCAL += TALENT:ARG:담력 * 10 + TALENT:ARG:성적흥미 * 10 - TALENT:ARG:자제심 * 10 - TALENT:ARG:무관심 * 10
DEBUGPRINTFORML 【DATE_EVENT({ARG})】DateEvent_Checksum = {LOCAL}
;헤어질 때에 선물
CALL GIFT(ARG)
IF RESULT == 1
;합의없음(키스 성공하지 않다)
ELSEIF !(CFLAG:ARG:1 & 1) && !(CFLAG:ARG:13 & 1)
	;성공 판정
	IF (LOCAL > 220 + TALENT:ARG:일선을넘지않음 * 30 + ABS(TALENT:ARG:키스미경험 * 20)) && CFLAG:ARG:합의판정 > 180 && (TALENT:ARG:사모 || TALENT:ARG:연모)
		CALL KOJO_MESSAGE_SEND("SP_EVENT", 1, ARG, 0)
		;캔슬 플래그
		IF !TCVAR:ARG:키스캔슬
			CALL ADD_KISS(ARG)
			SETBIT CFLAG:ARG:데이트후이벤트플래그 , 0
		ENDIF
	ELSEIF !MARK:ARG:반발각인
		CALL KOJO_MESSAGE_SEND("SP_EVENT", 3, ARG, 0)
	ENDIF
;반발각인 소거 처리
ELSEIF MARK:ARG:반발각인 && LOCAL > 50
	;ABL:친밀 LVUP 처리 혼잡. 해독이 귀찮았기 때문에 요소들들 취해 타…
	FOR LOCAL, 0,4
		주부족 = 0
		경험부족 = 0
		SIF JUEL_DEMAND(9, LOCAL, 0, ARG) == -2
			BREAK
		SIF JUEL_DEMAND(9, LOCAL, 0, ARG) < 0
			CONTINUE
		;필요한 보주를 찾는다
		FOR LOCAL:1, 0,100
			IF JUEL_DEMAND(9, LOCAL, LOCAL:1, ARG) > 0
				IF JUEL:(ARG):(LOCAL:1) < JUEL_DEMAND(9, LOCAL, LOCAL:1, ARG) / 2
					주부족 ++
					BREAK
				ENDIF
			ENDIF
		NEXT
		SIF 주부족
			CONTINUE
		;필요한 경험을 찾는다
		FOR LOCAL:1, 0,100
			IF EXP_DEMAND(9, LOCAL, LOCAL:1, ARG) > 0
				IF  EXP:(ARG):(LOCAL:1) < EXP_DEMAND(9, LOCAL, LOCAL:1, ARG) / 2
					경험부족 ++
					BREAK
				ENDIF
			ENDIF
		NEXT
		SIF 경험부족
			CONTINUE
		;여기까지 올 수 있으면(자) 반발각인 소거
		;요컨데 데이트로 좋은 분위기＋친밀 LVUP 수준이1/2채우면(자) 반발 소거
		MARK:ARG:반발각인 --
		CALL KOJO_MESSAGE_SEND("SP_EVENT", 3, ARG, (MARK:ARG:반발각인 + 1))
		PRINTL
		IF MARK:ARG:반발각인
			PRINTFORMW %CALLNAME:ARG%는 반발각인 LV{MARK:ARG:반발각인}에 내렸다
		ELSE
			PRINTFORMW %CALLNAME:ARG%는 반발각인 LV1를 잃었다
		ENDIF
		BREAK
	NEXT
;고백하고 있지 않다(MASTER에 연인이 없다)
ELSEIF !(CFLAG:ARG:1 & 4) && !(CFLAG:ARG:13 & 2) &&  !FLAG:고백금지 && CAN_MAKE_연인()
	;성공 판정
	IF (LOCAL > 530 + TALENT:ARG:일선을넘지않음 * 20)
		CALL KOJO_MESSAGE_SEND("SP_EVENT",2,ARG,0)
		IF FLAG:스침연인수 == 100 && TALENT:ARG:스침
			PRINTFORML 　- 포말의 사랑의 상대는 100명까지 밖에 기억할 수 없습니다
			PRINTFORML 　　받았을 경우, 지금까지의 누군가의 이름을 잊어 버립니다 -
		ELSEIF FLAG:추가연인수 > 0
			PRINTFORML 　- 연인은{1 + FLAG:추가연인수}인까지로 밖에 취득 할 수 없습니다 -
		ELSE
			;나올 수 없는 경우의 수
			PRINTFORML 　- 연인의 수에 제한은 없습니다 -
		ENDIF
		PRINTFORML %CALLNAME:ARG%의 고백을…
		CALL ASK_YN("받는다","거절한다")
		IF !RESULT
			SETBIT CFLAG:ARG:기정사실 , 2
			SETBIT CFLAG:MASTER:기정사실, 2
			TALENT:MASTER:연인 = 1
			CALL SET_연인(NO:ARG)
			SETBIT CFLAG:ARG:데이트후이벤트플래그 , 1
			CALL KOJO_MESSAGE_SEND("SP_EVENT",2,ARG,1)
			PRINTL
			PRINTFORMW %ARG와% 연인이 되었습니다
		ELSE
			CALL KOJO_MESSAGE_SEND("SP_EVENT",2,ARG,2)
		ENDIF
	ELSEIF !MARK:ARG:반발각인
		CALL KOJO_MESSAGE_SEND("SP_EVENT",3,ARG,0)
	ENDIF
ELSE
	CALL KOJO_MESSAGE_SEND("SP_EVENT",3,ARG,0)
ENDIF

@GIFT(ARG)
#DIM 호감도상승치
#DIM 선물평가
;데이트전호감도와 현호감 도를 비교
호감도상승치 = CFLAG:ARG:2 - TFLAG:데이트전호감도
IF DAY:2 == 4 && DAY:3 == 24 && ITEM:94 && TFLAG:데이트전호감도
	PRINTFORML 크리스마스선물을 건네줄까?
	CALL ASK_YN
	IF !RESULT
		PRINTFORMW %CALLNAME:ARG%에 선물을 건네주었다
		ITEM:94 = 0
		IF TALENT:ARG:연인
			PRINTFORMW %CALLNAME:ARG%의 얼굴이 순식간에 붉게 물들어 간다
			PRINTFORMW ……볼에 쪽 해 줄 수 있었다
			선물평가 = 10
		ELSEIF TALENT:ARG:연모
			PRINTFORMW %CALLNAME:ARG%의 얼굴이 순식간에 붉게 물들어 간다
			PRINTFORMW 선물을 받은 %CALLNAME:ARG%는 행복 그 자체라고 하는 느낌이다
			선물평가 = 7
		ELSEIF TALENT:ARG:사모
			PRINTFORML %CALLNAME:ARG%는 놀라면서도 기꺼이 받아 주었다
			PRINTFORML 스트레이트하게 호의를 나타난%CALLNAME:ARG%는, %마스터를% 한층 더 의식한 것 같다
			선물평가 = 5
		ELSE
			PRINTFORMW %CALLNAME:ARG%는 고가의 물건을 건네받았던 것에 당황해도 있는 것 같지만, 그다지도 아닌 것 같다
			선물평가 = 2
		ENDIF
		CFLAG:ARG:2 += 호감도상승치 * 선물평가 / 5
		CFLAG:ARG:4 += 호감도상승치 * 선물평가 / 10
		RETURN 1
	ELSE
		RETURN 0
	ENDIF
ELSE
	RETURN 0
ENDIF

@WITH_연인(ARG)
#FUNCTION
;연인의 몇 사람과 ARG가 같은 장소에 있을까를 돌려준다
VARSET LOCAL
SIF ARG == 0
	ARG = MASTER
FOR LOCAL,1,CHARANUM
	SIF CFLAG:LOCAL:현재위치 == CFLAG:ARG:현재위치 && TALENT:LOCAL:연인
		LOCAL:1 ++
NEXT
RETURNF LOCAL:1

@CAN_MAKE_연인()
#FUNCTION
;연인が今何人いるかカウント
VARSET LOCAL
FOR LOCAL, 1, CHARANUM
	SIF TALENT:LOCAL:연인
		LOCAL:1 ++
NEXT
;KR판 복수연인 대응용 주석처리
;SIF LOCAL:1 < (1 + FLAG:추가연인수)
;	RETURNF 1
;RETURNF 0
RETURNF 1
;===========================================
;키스경험やファーストキス処理などをまとめてやる関数
;===========================================
@ADD_KISS(ARG)
IF TALENT:MASTER:키스미경험
	CALL SET_FIRSTKISS(MASTER, ARG, 1)
	EXP:ARG:애정경험 += 10
ENDIF
IF TALENT:ARG:키스미경험
	CALL SET_FIRSTKISS(ARG, MASTER, 1)
	EXP:ARG:애정경험 += 10
ENDIF
EXP:ARG:키스경험 ++
EXP:MASTER:키스경험 ++
SETBIT CFLAG:ARG:기정사실 , 0


code:
@TURN_RESET
;처녀상실
FOR LOCAL, 0, CHARANUM
	SIF !TCVAR:LOCAL:파과
		CONTINUE
	IF TCVAR:LOCAL:파과 == 1
		IF TALENT:LOCAL:처녀 == 1
			TALENT:LOCAL:처녀 = 0
			EX:LOCAL:처녀상실체크 = 1
		ENDIF
		IF LOCAL && (FLAG:시간정지 || TCVAR:LOCAL:만취 || CFLAG:LOCAL:수면)
			TALENT:LOCAL:처녀 = -1
			EX:LOCAL:처녀상실체크 = 1
			CFLAG:LOCAL:292 = 1
		ENDIF
	ENDIF
	IF TCVAR:LOCAL:파과 == 2 && TALENT:LOCAL:처녀 == 2
		TALENT:LOCAL:처녀 = 0
		EX:LOCAL:처녀상실체크 = 2
	ENDIF
	IF TCVAR:LOCAL:파과 == -1 && TALENT:LOCAL:처녀 == -1
		TALENT:LOCAL:처녀 = 0
		EX:LOCAL:처녀상실체크 = 1
	ENDIF
NEXT

;동정상실
IF HAS_PENIS(MASTER)
	SELECTCASE SELECTCOM
		CASE 60, 61, 65, 67, 68, 320, 322
			TALENT:MASTER:1 |= 1
		CASE 62, 63, 66, 69, 70, 323
			TALENT:MASTER:1 |= 2
	ENDSELECT
ENDIF
IF HAS_PENIS(TARGET)
	SELECTCASE SELECTCOM
		CASE 64
			TALENT:1 |= 1
		CASE 92, 93, 94
			TALENT:1 |= 2
		CASE 71
			IF MASTER_POSE(SET_V, SET_P, 1)
				TALENT:1 |= 2
			ELSE
				TALENT:1 |= 1
			ENDIF
	ENDSELECT
ENDIF

;키스 성공
;합의 취득
SIF (SELECTCOM == 312 || SELECTCOM == 602) && TFLAG:193 != 99 && !CFLAG:MASTER:장난
	SETBIT CFLAG:기정사실, 0

;고백 성공
;연인 취득
IF SELECTCOM == 352 && TFLAG:193
	SETBIT CFLAG:기정사실, 2
	SETBIT CFLAG:MASTER:기정사실, 2
	TALENT:MASTER:연인 = 1
	CALL SET_연인(NO:TARGET)
	;섹프 삭제
	IF TALENT:TARGET:섹프 == 1
		TALENT:TARGET:섹프 = 0
	ENDIF
ENDIF
;; 프로포즈 성공
IF SELECTCOM == 380 && TFLAG:193
	SETBIT CFLAG:기정사실, 5
	SETBIT CFLAG:MASTER:기정사실, 5
	SETBIT TALENT:연인,1
	CALL SET_아내(NO:TARGET)
	ITEM:약혼반지=0
	FLAG:엔딩여부=1
	;섹프 삭제
	IF TALENT:TARGET:섹프 == 1
		TALENT:TARGET:섹프 = 0
	ENDIF
ENDIF

;식사
SIF SELECTCOM == 414 || SELECTCOM == 415
	CALL RESET_DISH

;조교 이력을 1 턴 보존
FOR LOCAL, 0, CHARANUM
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	TCVAR:LOCAL:110 = TCVAR:LOCAL:10
	TCVAR:LOCAL:112 = TCVAR:LOCAL:12

	SIF TCVAR:LOCAL:102
		TCVAR:LOCAL:102 --
	SIF !TCVAR:LOCAL:102
		TCVAR:LOCAL:103 = -1
	SIF TCVAR:LOCAL:104
		TCVAR:LOCAL:104 --
	SIF !TCVAR:LOCAL:104
		TCVAR:LOCAL:105 = -1
	;체위의 보존
	FOR LOCAL:1, 1, 11
		TEQUIP:LOCAL:(110 + LOCAL:1) = TEQUIP:LOCAL:(100 + LOCAL:1)
	NEXT
	;몸의자세제어
	IF TCVAR:LOCAL:몸의자세제어
		TCVAR:LOCAL:몸의자세제어 --
		SIF TCVAR:LOCAL:몸의자세제어 == 0
			TCVAR:LOCAL:몸의자세 = 0
	ENDIF
	;위치관계제어
	IF TCVAR:LOCAL:위치관계제어
		TCVAR:LOCAL:위치관계제어 --
		SIF TCVAR:LOCAL:위치관계제어 == 0
			TCVAR:LOCAL:위치관계 = 0
	ENDIF
	;내친 김에 사정 처리용 캐릭터라인 변수의 초기화
	CSTR:LOCAL:10 = /
	CSTR:LOCAL:11 = /
	;복장 플래그
	IF FLAG:시간정지 == 1
	ELSE
		CALL CLOTHES_SETTING_TRAIN(LOCAL)
	ENDIF
	IF BATHROOM(CFLAG:LOCAL:현재위치) && GET_MAPID(CFLAG:LOCAL:현재위치) == MAIN_MAP
		;목욕탕 플래그는 MOVEMENT공통처리에 이동했습니다
;		CFLAG:LOCAL:목욕탕 = 0
		RESET_STAIN LOCAL
		CFLAG:LOCAL:질내사정 = 0
		CFLAG:LOCAL:애널사정 = 0
		CFLAG:LOCAL:구내정액 = 0
		CFLAG:LOCAL:안면정액 = 0
		CFLAG:LOCAL:손에정액 = 0
	ELSEIF CFLAG:LOCAL:목욕탕 > 120 && !CFLAG:LOCAL:우후후
		STAIN:LOCAL:4 |= 더러운_애널
	ENDIF
NEXT
;BASE의 변동
FOR LOCAL, 0, CHARANUM
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	;발기도
	;가산 후
	CFLAG:LOCAL:발기도2 = BASE:LOCAL:발기
	BASE:LOCAL:발기 = MIN(MAXBASE:LOCAL:발기, BASE:LOCAL:발기)
	IF NOWEX:LOCAL:사정
		IF !MASTER_POSE(SET_V, SET_P) && !MASTER_POSE(SET_A, 1)
			BASE:LOCAL:발기 = BASE:LOCAL:발기 * (ABL:LOCAL:욕망 + 1) * (BASE:LOCAL:정력 + 200) / ((MAXBASE:LOCAL:정력 + 200) * (ABL:LOCAL:욕망 + 10))
		ELSE
			BASE:LOCAL:발기 = 1000
		ENDIF
	ENDIF
NEXT

SIF CFLAG:MASTER:우후후
	CALL 우후후안특수이벤트

;플래그의 리셋
TFLAG:192 = 0
TFLAG:193 = 0
TFLAG:194 = 0
TSTR:2 =

;TCVAR의 리셋
FOR LOCAL, 0, CHARANUM
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	FOR LOCAL:1, 0, 100
		TCVAR:LOCAL:(LOCAL:1) = 0
	NEXT
;삽입 이력
	TCVAR:LOCAL:100 = TEQUIP:LOCAL:50
	TCVAR:LOCAL:101 = TEQUIP:LOCAL:51
NEXT

TFLAG:110 = TCVAR:MASTER:친해져

;FIRSTTIME에 SELECTCOM를 처넣는다
CALLF FIRSTTIME(SELECTCOM)
FOR LOCAL, 1, CHARANUM
	SIF EXP_UP(15, LOCAL)
		CALLF FIRSTTIME("정음", 0, LOCAL)
NEXT
;사정장소
FOR LOCAL:1, 0, 60
	IF GETBIT(TFLAG:1, LOCAL:1)
		LOCALS = "사정"{LOCAL:1}
		CALLF FIRSTTIME(LOCALS, 0, TARGET)
	ENDIF
NEXT
PREVCOM = SELECTCOM
;파생 커맨드의 이력
TFLAG:151 = TFLAG:50
;전회의 TARGET
IF TFLAG:전회커맨드결과 > 0
	TFLAG:103 = TARGET
	TFLAG:150 = ASSIPLAY
ENDIF

;경험치취표시용
FOR LOCAL, 0, CHARANUM
	SIF CFLAG:LOCAL:출금
		CONTINUE
	;현재위치의 보존
	CFLAG:LOCAL:전턴위치 = CFLAG:LOCAL:현재위치
	FOR LOCAL:1, 0, 100
		TCVAR:LOCAL:(400 + LOCAL:1) = EXP:LOCAL:(LOCAL:1)
	NEXT
NEXT
IF BATHCHECK(CFLAG:MASTER:현재위치) && !FLAG:시간정지 && AT_HOME(MASTER)
	FOR LOCAL, 1, CHARANUM
		SIF 목욕탕추방조건(LOCAL) && CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
			CALL 목욕탕으로부터내쫒아(LOCAL)
	NEXT
ENDIF
;TFLAG의 리셋
VARSET TFLAG, 0, 0, 100
TFLAG:400 = 0
TFLAG:이동불능메시지 = 0
;졸리면 기력이 감소해 나간다
IF !CFLAG:MASTER:수면
	IF NEMUKE() >= 960
		BASE:MASTER:체력 = MAX(BASE:MASTER:체력 - 5 * TIME_PROGRESS(TIME:1) , 0)
		BASE:MASTER:기력 = MAX(BASE:MASTER:기력 - 10 * TIME_PROGRESS(TIME:1) , 0)
	ELSEIF NEMUKE() >= 840
		BASE:MASTER:기력 = MAX(BASE:MASTER:기력 - 2 * TIME_PROGRESS(TIME:1) , 0)
	ENDIF
ENDIF

@우후후안특수이벤트
#DIM 탈주도권확률
VARSET LOCAL
;당신 주도
IF CFLAG:TARGET:우후후 == 1 && CFLAG:욕구불만도 > 300 && !TCVAR:TARGET:억지로
;공수 역전
	IF SHIRAHU(TARGET) && TEQUIP:TARGET:체위 == 3
		;강할 만큼(정도) 확률이 높은→분모는 작다
		;1/확률
		탈주도권확률 = 15
		탈주도권확률 -= TALENT:TARGET:담력 * 2
		탈주도권확률 -= TALENT:TARGET:응답 * 2
		탈주도권확률 -= TALENT:TARGET:자존심 * 2
		탈주도권확률 += TALENT:TARGET:자제심 * 3
		탈주도권확률 -= TALENT:TARGET:성적흥미 * 3
		탈주도권확률 += TALENT:TARGET:수치심 * 2
		탈주도권확률 -= TALENT:TARGET:새드 * 5
		탈주도권확률 += TALENT:TARGET:마조 * 5

		;RAND에 건네주므로 최저치는 1
		SIF 탈주도권확률 < 1
			탈주도권확률 = 1
		IF !RAND(탈주도권확률)
			CFLAG:MASTER:우후후 = 2
			TFLAG:COMABLE관리 = 3
			FOR LOCAL, 1, CHARANUM
				IF CFLAG:MASTER:현재위치 == CFLAG:LOCAL:현재위치
					CFLAG:LOCAL:우후후 = 2
					LOCAL:1 ++
				ENDIF
			NEXT
			PRINTFORML 깨달으면(자), %CALLNAME:TARGET%\@ LOCAL:1 > 1 ?  들 #  \@에게 주도권을 빼앗기고 있었다
		ENDIF
	ENDIF
;상대 주도
ELSEIF CFLAG:TARGET:우후후 == 2
	IF BASE:기력 <= 0
		PRINTFORML %타겟은% 조금 지쳤다고 해 %마스터를% 해방했다
		CALL 덮쳐지기종료("지쳤다")
	ELSEIF CFLAG:욕구불만도 <= 0 && !TCVAR:TARGET:마음대로해
		PRINTFORML %타겟은% 만족한 것 같다
		CALL KIGEN_CHANGE(TARGET, 100,1)
		BASE:분노 = 0
		CFLAG:언짢음 = 0
		CALL ASK_M("빠져 나가는", 1,"여기는 아직 만족하고 있지 않는", 1,"좀 더 공격해", ABL:새드끼 >= 3)
		SELECTCASE RESULT
			CASE 0
				PRINTFORML %타겟은% %마스터를% 해방했다
				CALL 덮쳐지기종료("만족")
			CASE 1
				PRINTFORML %마스터는% %타겟을% 덮치기
				CALL KOJO_MESSAGE_SEND("EVENT", 30, TARGET, 3)
				FOR LOCAL, 0, CHARANUM
					CFLAG:(TARGET:LOCAL):우후후 = 1
					IF INROOM(CFLAG:MASTER:현재위치) == 2
						CALL DATUI_SHOES(LOCAL, 0)
						CALL DATUI_SHOES(MASTER, 0)
					ENDIF
				NEXT
			CASE 2
				PRINTFORML %타겟은% 웃음을 띠며 그렇다면 망가질 때까지 해 주면(자) 속삭였다
				CALL KOJO_MESSAGE_SEND("EVENT", 30, TARGET, 4)
				TCVAR:TARGET:마음대로해 = 1
		ENDSELECT
	ENDIF
ENDIF

@목욕탕추방조건(ARG)
#FUNCTION
SIF FLAG:70
	RETURNF 0
SIF !GETBIT(TALENT:MASTER:2, 1) && HETEROSEX(MASTER, ARG)
	RETURNF 0
SIF TALENT:ARG:수치심 < 0 && ABL:ARG:친밀 >= 5
	RETURNF 0
SIF CFLAG:ARG:우후후
	RETURNF 0
SIF CFLAG:ARG:태도 < 2
	RETURNF 1
SIF CFLAG:ARG:기정사실 < 1
	RETURNF 1
SIF !(TALENT:ARG:연모 || TALENT:ARG:섹프)
	RETURNF 1

RETURNF 0


code:
;FileName_EVENTCOMEND.ERB -------------------------- Rev1. 00
;턴 종료의 처리
;CALL		SYSTEM
;ARG		VOID
;RETURN		VOID
;COMMENT
;-----------------------------------------------------------
@EVENTCOMEND
#DIM 장비부위
#DIMS 아이의이름
;일동작의 에러 체크
;FOR LOCAL, 0, CHARANUM
;	SIF CFLAG:LOCAL:326 == 0 && CFLAG:LOCAL:행동 == 5
;		PRINTFORMW ### COM_END 326ERROR %CALLNAME:LOCAL% ### CHARA_NO:{LOCAL}
;	SIF CFLAG:LOCAL:현재위치 == 0
;		PRINTFORMW ### COM_END LOCATION ERROR %CALLNAME:LOCAL% 의 현재위치가 이상합니다 ### CHARA_NO:{LOCAL}
;NEXT
;턴 종료 공통 처리
CALL TURN_RESET

SIF FLAG:연회개최플래그 > 0
	CALL ENKAI_SINKOU

SIF TFLAG:102 < 2 && TARGET && SHIRAHU(TARGET) && !FLAG:70
	CALL 봇슈트
IF BASE:MASTER:술기운 >= MAXBASE:MASTER:술기운
	;AT_HOME 보는 것만으로도 좋을지도 모른다
	IF !AT_HOME(MASTER) && GET_MAPID(CFLAG:MASTER:현재위치) != MAIN_MAP
		CALL 데이트종료타임업처리
	ELSE
		PRINTFORMW 한계까지 계속 마신 %마스터는%, 그대로 딱과 넘어져 버렸다…
	ENDIF
	GOTO OYASUMI
ENDIF
IF (BASE:MASTER:체력 <= 500 && !TCVAR:TARGET:마음대로해) || BASE:MASTER:체력 <= 0
	PRINT (체력이 한계에 와 있습니다)
	;밀어넘어뜨려지고 안이라면 분기
	IF TFLAG:102 == 3 && BASE:MASTER:체력 > 0 && !TCVAR:TARGET:마음대로해
		PRINTL
		PRINTL [0]마음대로해 주는 [1]해방받는다
		INPUT
		IF !RESULT
			TCVAR:TARGET:마음대로해 = 1
			BASE:MASTER:체력 -= 250
			GOTO SUKINISITE
		ENDIF
	ENDIF
	PRINTW 　하루를 종료합니다
	IF BASE:MASTER:체력 <= 0 && !CHK_DATENOW(CFLAG:MASTER:데이트중)
		PRINTFORMW %마스터는% 체력을 다 써 버려 실신해 버렸다…
		FLAG:기절중단 = 100 + CFLAG:MASTER:현재위치 / 100
	ENDIF
	;밀어넘어뜨려지고 안이라면 TARGET에 기학 경험
	IF TFLAG:102 == 3
		IF BASE:MASTER:체력 <= 0
			EXP:가학쾌락경험 += 10
			PRINTL
			PRINTFORMW 가학쾌락경험 +10(%CALLNAME%)
			PRINTL
		ELSE
			EXP:가학쾌락경험 += 2
			PRINTL
			PRINTFORMW 가학쾌락경험 +2(%CALLNAME%)
			PRINTL
		ENDIF
	ENDIF
	$OYASUMI
	CALL SET_SLEEP(1, MASTER, 0)
	CFLAG:MASTER:취침시간 = DAY * 1440 + TIME
	;외박 할 수 있는 상태라면 그대로 잔다(연모 전제)
	IF CFLAG:MASTER:현재위치 == OMANEKIBEYA() && !TCVAR:(CFLAG:MASTER:초대):연속숙박금지
		IF CFLAG:(CFLAG:MASTER:초대):데이트중
			PRINTFORMW 오늘은 %CALLNAME:(CFLAG:MASTER:초대)%가 묵게 해 주는 것 같다
			FLAG:기절중단 = 0
		ENDIF
	ELSEIF TARGET > 0 && TALENT:TARGET:연모 && BASE:TARGET:체력 >= 500 && !CFLAG:TARGET:수면 && !CFLAG:MASTER:동침중 && ((CFLAG:TARGET:동행 && CHK_DATENOW(CFLAG:MASTER:데이트중)) || !CHK_DATENOW(CFLAG:MASTER:데이트중))
		IF CHK_DATENOW(CFLAG:MASTER:데이트중)
			PRINTFORMW %CALLNAME:TARGET%에 의지하고 %마스터는% %GET_MAPNAME(MAIN_MAP)%까지 돌아왔다
		ENDIF
		IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
			PRINTFORMW %타겟이% %마스터를%사실로 옮겨 주었다…
			CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
			CALL KOJO_MESSAGE_SEND("EVENT", 3, TARGET, 6,0)
		ELSE
			PRINTFORMW %타겟이% %마스터를%\@ INROOM(CFLAG:MASTER:현재위치) == 2 ?  침대 # 이불 \@로 옮겨 주었다…
		ENDIF
		CFLAG:MASTER:데이트중 = MAIN_MAP
		FLAG:기절중단 = 0
	ELSEIF CFLAG:MASTER:현재위치 == 난쟁이벌레통
		CFLAG:MASTER:현재위치 = 툇마루
		PRINTFORMW 어느 새인가 %마스터는% 툇마루에 널려 있었다…
	ELSEIF CHK_DATENOW(CFLAG:MASTER:데이트중)
		PRINTFORMW %마스터는% 마지막 힘을 떨쳐 짜, %GET_MAPNAME(MAIN_MAP)%까지 돌아왔다
		FLAG:기절중단 = 100 + CFLAG:MASTER:현재위치 / 100
		CFLAG:MASTER:현재위치 = ENTRANCE
		PRINTFORMW
	ELSE
		PRINTFORMW %마스터는% 휘청거리는 발걸음으로,사실로 돌아왔다
		CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
		FLAG:기절중단 = 0
	ENDIF
	FOR LOCAL, 0, CHARANUM
		SIF CFLAG:LOCAL:출금
			CONTINUE
		CALL ENDUFUFU(LOCAL)
		CALL SET_PRANK(0, LOCAL)
		CFLAG:LOCAL:데이트중 = MAIN_MAP
		CFLAG:LOCAL:동행 = 0
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		SIF !BATHCHECK(CFLAG:LOCAL:현재위치) && TEQUIP:LOCAL:목욕탕플레이
			TEQUIP:LOCAL:목욕탕플레이 = 0
	NEXT
;	수치 변수가 소수점 이하를 잘라 버리는 특성을 악용 한 것이 이쪽, 다른 변수를 사용하지 않는 분 이쪽 쪽이 좋은 것인지도?
	CFLAG:MASTER:데이트중 = CFLAG:MASTER:현재위치 / 100
	TFLAG:106 = 0
	BEGIN AFTERTRAIN
ENDIF
IF TARGET && TFLAG:키스마크 == TARGET && TFLAG:붕대 && !RAND:2
	PRINTFORMW %CALLNAME:MASTER%의 목덜미를 가리는 것을 알아차린%타겟은%, 붕대를 인가 했다
	TFLAG:붕대 = 0
	BASE:분노 += 500
ENDIF
$SUKINISITE
IF TARGET && CFLAG:언짢음 && !FLAG:70
	IF TALENT:TARGET:담력 < 0
		PRINTFORMW %타겟을% 울려 버렸다…
		CALL KOJO_MESSAGE_SEND("EVENT", 29, TARGET, 2,0)
	ELSE
		PRINTFORMW %타겟을% 화나게 해 버렸다…
		CALL KOJO_MESSAGE_SEND("EVENT", 29, TARGET, 1,0)
	ENDIF

	IF !TCVAR:억지로 && FLAG:모리야복권데이트상대 != TARGET
		FOR LOCAL, 0, CHARANUM
			CALL ENDUFUFU(LOCAL)
		NEXT
		TFLAG:106 = 0
	ENDIF
ENDIF

IF FLAG:70 == 1
	IF BASE:MASTER:TSP <= 0
		PRINTFORML (TSP가 없어졌습니다)
		[SKIPSTART]
		FOR LOCAL, 1, CHARANUM
			IF CFLAG:LOCAL:현재위치 == CFLAG:PLAYER:현재위치
				FOR 장비부위, 0,23
					SIF (!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위))
						LOCAL:1 ++
					SIF 장비부위 == 6 && CFLAG:LOCAL:노팬티 && ((!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위)))
						LOCAL:1 --
				NEXT
				IF LOCAL:1
					EQUIP:LOCAL:0 = 1
					IF CFLAG:LOCAL:시간정지착의 == 1
						PRINTFORMW %마스터는% %CALLNAME:LOCAL%를\@ CFLAG:LOCAL:노팬티 ?  팬티 이외를 #  \@원래 모습에 되돌렸다!
						CFLAG:LOCAL:시간정지착의 = 0
					ELSE
						PRINTFORMW %마스터는% %CALLNAME:LOCAL%에\@ CFLAG:LOCAL:노팬티 ?  팬티 이외의#  \@옷을 입혔다
					ENDIF
				ENDIF
			ENDIF
		NEXT
		[SKIPEND]
		IF CFLAG:MASTER:현재위치 != OMANEKIBEYA()
			PRINTFORMW %마스터는% 당황해 그 자리를 떨어졌다
			TFLAG:운반 = 0
			CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
			FOR LOCAL:10, 0, CHARANUM - 1
				LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
				SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
					CONTINUE
				IF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 30
					PRINTFORMW 돌연 격렬한 절정에 습격당한 %CALLNAME:LOCAL%는, 단 교성을 주어 그 자리에 거치거나 붐비어 버렸다
					CFLAG:LOCAL:333 += 20
				ELSEIF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 0
					PRINTFORMW %CALLNAME:LOCAL%는, 수수께끼의 쾌감에 깜짝과 몸을 경직시켰다
				ELSEIF CFLAG:LOCAL:291 == 1
					PRINTFORMW 알지 않는 사이에 팬티를 탈의한 %CALLNAME:LOCAL%는 침착하고 없을 것 같다
				ENDIF
				SIF EX:LOCAL:처녀상실체크 == 1
					PRINTFORMW %CALLNAME:LOCAL%는 고간의 둔통에 얼굴을 찡그리고 있는…
			NEXT
			IF !AT_HOME(MASTER)
				PRINTFORML %GET_MAPNAME(MAIN_MAP)%까지 전력 질주 해 버렸다…
				BASE:MASTER:체력 -= TIME_REQUIRED(MAIN_MAP, CFLAG:MASTER:데이트중) * 5
				BASE:MASTER:기력 -= TIME_REQUIRED(MAIN_MAP, CFLAG:MASTER:데이트중) * 5
				BASE:MASTER:체력 = MAX(BASE:MASTER:체력, 0)
				BASE:MASTER:기력 = MAX(BASE:MASTER:기력, 0)
				TIME += TIME_REQUIRED(MAIN_MAP, CFLAG:MASTER:데이트중) / 5
			ENDIF
			IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
				CFLAG:MASTER:현재위치 = CFLAG:MASTER:초기위치
			ELSE
				CFLAG:MASTER:현재위치 = 피난처
			ENDIF
			BASE:MASTER:TSP = 0
			CALL AFTER_AFFAIR
			FOR LOCAL, 0, CHARANUM
				IF CFLAG:LOCAL:우후후 == 1 && LOCAL != 0
					;우후후 종료 말호출
					CALL KOJO_MESSAGE_SEND("EVENT", 15, LOCAL, 2,0)
				ENDIF
				CALL ENDUFUFU(LOCAL)
				CFLAG:LOCAL:데이트중 = MAIN_MAP
			NEXT
			FLAG:데이트상대 = 0
			TFLAG:데이트전호감도 = 0
			TFLAG:106 = 0
		ELSE
			CALL FISHER_YATES_SHAFFLE(CHARANUM - 1)
			FOR LOCAL:10, 0, CHARANUM - 1
				LOCAL = SHAFFLE_ARRAY:(LOCAL:10) + 1
				IF CFLAG:LOCAL:현재위치 == CFLAG:MASTER:현재위치
					IF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 30
						PRINTFORMW 돌연 격렬한 절정에 습격당한 %CALLNAME:LOCAL%는, 단 교성을 주어 그 자리에 거치거나 붐비어 버렸다
						CFLAG:LOCAL:333 += 20
					ELSEIF EX:LOCAL:C절정 + EX:LOCAL:V절정 + EX:LOCAL:A절정 + EX:LOCAL:B절정 + EX:LOCAL:M절정 > 0
						PRINTFORMW %CALLNAME:LOCAL%는, 수수께끼의 쾌감에 깜짝과 몸을 경직시켰다
					ELSEIF CFLAG:LOCAL:291 == 1
						PRINTFORMW 알지 않는 사이에 팬티를 탈의한 %CALLNAME:LOCAL%는 침착하고 없을 것 같다
					ENDIF
					SIF EX:LOCAL:처녀상실체크 == 1
						PRINTFORMW %CALLNAME:LOCAL%는 고간의 둔통에 얼굴을 찡그리고 있는…
				ENDIF
				CALL ENDUFUFU(LOCAL)
			NEXT
			CFLAG:MASTER:우후후 = 0
		ENDIF
		FLAG:70 = 0
		RESETBGCOLOR
		BASE:MASTER:TSP = 0
	ENDIF
	SIF CFLAG:MASTER:우후후 && BASE:MASTER:발기 <= 900 && BASE:MASTER:정력 >= 500
		BASE:MASTER:발기 += 10
ENDIF
IF TARGET && BASE:체력 < 500
	IF FLAG:70 == 1 && CFLAG:MASTER:현재위치 != OMANEKIBEYA()
		IF !CFLAG:쇠약
			PRINTFORML (%CALLNAME:TARGET%의 체력이 한계에 와 있습니다)
			PRINTFORMW %마스터는% 그 자리를 떨어져, 시간정지를 해제했다
			CFLAG:쇠약 = 360
			FLAG:70 = 0
			FLAG:데이트상대 = 0
			TFLAG:데이트전호감도 = 0
			RESETBGCOLOR
			IF AT_HOME(MASTER)
				CFLAG:MASTER:데이트중 = MAIN_MAP
				IF CFLAG:MASTER:현재위치 != CFLAG:MASTER:초기위치
					CFLAG:MASTER:현재위치 = CFLAG:MASTER:초기위치
				ELSE
					CFLAG:MASTER:현재위치 = 피난처
				ENDIF
			ELSE
				IF STR:(6000 + CFLAG:MASTER:현재위치 + 10) == ""
					CFLAG:MASTER:현재위치 -= 10
				ELSE
					CFLAG:MASTER:현재위치 += 10
				ENDIF
			ENDIF
		ENDIF
		CALL CHARA_SLEEP, TARGET, 0
		FOR LOCAL, 0, CHARANUM
			CALL ENDUFUFU(LOCAL)
			CALL SET_PRANK(0, LOCAL, 1)
			SIF LOCAL
				CFLAG:LOCAL:데이트중 = MAIN_MAP
			CFLAG:LOCAL:동행 = 0
		NEXT
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		TFLAG:106 = 0
	ELSE
		IF !CFLAG:쇠약
			PRINTFORMW (%CALLNAME:TARGET%의 체력이 한계에 와 있습니다)
			CFLAG:쇠약 = 360
			FLAG:70 = 0
			RESETBGCOLOR
		ENDIF
		CALL CHARA_SLEEP, TARGET, 0
		FOR LOCAL, 0, CHARANUM
			CALL ENDUFUFU(LOCAL)
			CALL SET_PRANK(0, LOCAL, 1)
			SIF LOCAL
				CFLAG:LOCAL:데이트중 = MAIN_MAP
			CFLAG:LOCAL:동행 = 0
		NEXT
		FLAG:데이트상대 = 0
		TFLAG:데이트전호감도 = 0
		TFLAG:106 = 0
	ENDIF
ENDIF

IF FLAG:자식묘사
	FOR LOCAL, 1, CHARANUM
		SIF !TALENT:LOCAL:육아중
			CONTINUE
		SIF CFLAG:LOCAL:현재위치 != CFLAG:MASTER:현재위치
			CONTINUE
		아이의이름 = %CHILDNAME:LOCAL:(TALENT:LOCAL:육아중)%
		;욕구 불만도 억제와 유아기의 공복 판정과 기상 판정
		IF TCVAR:LOCAL:아이공복도 != -1
			SIF TCVAR:LOCAL:아이공복카운트
				TCVAR:LOCAL:아이공복도 += (TIME:0 - TCVAR:LOCAL:아이공복카운트)
			TCVAR:LOCAL:아이공복카운트 = TIME:0
			SIF CFLAG:LOCAL:욕구불만도 > 300
			CFLAG:LOCAL:욕구불만도 = 300
		ELSEIF TCVAR:LOCAL:아이공복도 == -1 && TIME:0 - TCVAR:LOCAL:아이공복카운트 > 120 && !RAND:2 && !CFLAG:MASTER:우후후 && CFLAG:LOCAL:출산경과일 < CHILD_어휘
			PRINTFORMDL %아이의이름%는 눈을 떴다
			TCVAR:LOCAL:아이공복도 = 120
			TCVAR:LOCAL:아이공복카운트 = TIME:0
			SIF CFLAG:LOCAL:욕구불만도 > 300
			CFLAG:LOCAL:욕구불만도 = 300
		ENDIF
		;통학
		IF GO_SCHOOL(LOCAL)
			IF !CHK_DATENOW(CFLAG:LOCAL:데이트중) && INRANGE(TIME, 420, 540) && TCVAR:LOCAL:아이공복도 != -1 && !CFLAG:MASTER:우후후
				PRINTFORMDL %아이의이름%는 서당으로 향했습니다
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_등하교")
				TCVAR:LOCAL:아이공복도 = -1
			;우후후중 돌아오지 않는 공기가 읽을 수 있는 아이
			ELSEIF !CHK_DATENOW(CFLAG:LOCAL:데이트중) && INRANGE(TIME, 840, 1020) && TCVAR:LOCAL:아이공복도 == -1 && !CFLAG:MASTER:우후후
				PRINTFORMDL %아이의이름%가 서당으로부터 돌아왔습니다
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 1, 0, "CHILD_RAISING_등하교")
				TCVAR:LOCAL:아이공복도 = 0
			ELSEIF INRANGE(TIME, 540, 840)
				TCVAR:LOCAL:아이공복도 = -1
			ELSEIF INRANGE(TIME, 1020, 1260)
				TCVAR:LOCAL:아이공복도 = 0
			ENDIF
		ENDIF
		;이 이후는 통학/수면중에는 발생하지 않는다
		SIF TCVAR:LOCAL:아이공복도 == -1
			CONTINUE
		;우후후가 중단하는 것 같은 이벤트도 만들어야할 것인가?
		SIF CFLAG:MASTER:우후후
			CONTINUE
		;신부가 자기도 하고 있으면(자) 아이도 무반응은의도
		SIF !SHIRAHU(LOCAL)
			CONTINUE
		;유아기
		IF CFLAG:LOCAL:출산경과일 < CHILD_어휘
			IF BATHROOM(CFLAG:현재위치) && !RAND:2
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_BATH", "유아기")
			ELSEIF TIME:0 <= CFLAG:LOCAL:취침시간 && TCVAR:LOCAL:아이공복도 >= 120 && CFLAG:LOCAL:출산경과일 < CHILD_젖떼기
					TCVAR:LOCAL:아이공복도 = 0
					EXP:LOCAL:분유경험 ++
					PRINTFORMDL %CALLNAME:LOCAL%는%아이의이름%에 수유를 하고 있다
					CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_식사", "수유")
			ELSEIF TIME:0 <= CFLAG:LOCAL:취침시간 && TCVAR:LOCAL:아이공복도 >= 120 && GROUPMATCH(SELECTCOM, 414, 415, 610, 615, 617)
					TCVAR:LOCAL:아이공복도 = 0
					PRINTFORMDL %CALLNAME:LOCAL%는%아이의이름%에 이유식을 주고 있다
					CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_식사", "이유식")
			ELSEIF TCVAR:LOCAL:아이공복도 < 120 && !RAND:6
				PRINTFORMDL %아이의이름%는 새근새근 자고 있다……
				TCVAR:LOCAL:아이공복도 = -1
					CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_SLEEPING", "유아기")
			ELSEIF CFLAG:LOCAL:출산경과일 >= CHILD_완구 && !RAND:5
				PRINTFORMDL %아이의이름%는 %마스터가% 만든 완구로 놀고 있다
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_TOY", "유아기")
			ELSEIF !RAND:3
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_OTHER", "유아기")
			ENDIF
		;유소년기
		ELSEIF INRANGE(CFLAG:LOCAL:출산경과일, CHILD_어휘, CHILD_자립전 -1)
			IF 아이의수면시간(LOCAL)
				PRINTFORMDW %아이의이름%는 침실로 향하고 있었던……
				TCVAR:LOCAL:아이공복도 = -1
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_SLEEPING", "유소년기")
			ELSEIF BATHROOM(CFLAG:현재위치) && !RAND:2
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_BATH", "유소년기")
			ELSEIF GROUPMATCH(SELECTCOM, 414, 415, 610, 615, 617)
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_식사", "유소년기")
			ELSEIF !RAND:6
				PRINTFORMDL %아이의이름%는 %마스터가% 만든 완구로 놀고 있다
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_TOY", "유소년기")
			ELSEIF !RAND:3
				TCVAR:LOCAL:아이공복도 = 0
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_OTHER", "유소년기")
			ENDIF
		;자립전
		ELSE
			;아이의 자립일이 아니면 잔다
			IF 아이의수면시간(LOCAL) && CFLAG:LOCAL:출산경과일 != CHILD_자립
				PRINTFORMDW %아이의이름%는 침실로 향하고 있었던……
				TCVAR:LOCAL:아이공복도 = -1
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_SLEEPING", "자립전")
			ELSEIF BATHROOM(CFLAG:현재위치) && !RAND:2
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_BATH", "자립전")
			ELSEIF GROUPMATCH(SELECTCOM, 414, 415, 610, 615, 617)
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_식사", "자립전")
			ELSEIF CFLAG:LOCAL:현재위치 == CFLAG:LOCAL:초기위치 && !RAND:8
				PRINTFORMDL %아이의이름%는 %CALLNAME:MASTER%에 만들어 준 나무의  완구를 추억 깊은 듯이 바라보고 있다……
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_TOY", "자립전")
			ELSEIF !RAND:5
				TCVAR:LOCAL:아이공복도 = 0
				CALL KOJO_MESSAGE_SEND("EVENT", 0, LOCAL, 0, 0, "CHILD_RAISING_OTHER", "자립전")
			ENDIF
		ENDIF
	NEXT
	RESETCOLOR
ENDIF

IF GET_EXISTMOB(MAIN_MAP)
	FOR LOCAL, MINROOM(), MAXROOM
		IF ROOMDATA:(LOCAL % 100) & 장소_모브
			YOGORE:LOCAL += TIME_PROGRESS(TIME:1) * MOB_YOGORE(MAIN_MAP)
			SIF CFLAG:MASTER:현재위치 == LOCAL && AT_HOME(MASTER)
				TRYCALLFORM MOB_TURNENDEVENT_{MAIN_MAP}
		ENDIF
	NEXT
ENDIF
IF !FLAG:70 && TARGET
	IF TALENT:TARGET:동물귀 && CFLAG:TARGET:900 == 7 && !TCVAR:TARGET:발정 && PALAM:TARGET:욕정 >= PALAMLV:5 && !TALENT:TARGET:임신
		PRINTFORML %타겟은% 뺨을 상기시켜, 숨을 할딱이며 있다…
		PRINTFORML 위험일에 극도로 흥분한 %타겟은% [발정]해 버린 것 같다
		TCVAR:TARGET:발정 = 1
	ENDIF
ENDIF

FOR LOCAL, 0, CHARANUM
	;IF !CFLAG:LOCAL:우후후
		FOR LOCAL:2, 101,109
			SIF TEQUIP:LOCAL:(LOCAL:2) != TEQUIP:LOCAL:(LOCAL:2 + 10)
				TEQUIP:LOCAL:(LOCAL:2) = 0
		NEXT
	;ENDIF
	SIF CFLAG:LOCAL:현재위치 == SUKIMA()
		CONTINUE
	;의상의 갈아입음 수선은@SHOW_STATUS에 이동
	[SKIPSTART]
	LOCAL:1 = 0
	FOR 장비부위, 0, 23
		SIF (!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위))
			LOCAL:1 ++
		SIF 장비부위 == 6 && CFLAG:LOCAL:노팬티 && ((!수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(200 + 장비부위)) || (수면시간(LOCAL) && EQUIP:LOCAL:장비부위 != CFLAG:LOCAL:(240 + 장비부위)))
			LOCAL:1 --
	NEXT
	IF !CFLAG:LOCAL:우후후 && CFLAG:LOCAL:현재위치 != BATHROOM(CFLAG:LOCAL:현재위치) && LOCAL:1
		IF 수면시간(LOCAL)
			;파자마 세트
			CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_파자마")
			CFLAG:LOCAL:파자마 = 1
		ELSE
			;평상복 세트
			CALL CTRL_CLOTHES_SET(LOCAL, "현재 의상의 변경_평상복")
			CFLAG:LOCAL:파자마 = 0
		ENDIF
		SIF !수면시간(LOCAL) && CFLAG:LOCAL:면간발각
			CFLAG:LOCAL:면간발각 = 0
		CALL CLOTHES_SETTING_TRAIN(LOCAL)
	ENDIF
	[SKIPEND]
	;일단 CLOTHES_SETTING_TRAIN는 해 손해는 없을 것이다
		SIF !수면시간(LOCAL) && CFLAG:LOCAL:면간발각
			CFLAG:LOCAL:면간발각 = 0
		CALL CLOTHES_SETTING_TRAIN(LOCAL)
NEXT
;강제적으로 WAIT가 추가되는 것을 회피하기 위해서 TWAIT를 가르친다
;이것으로 입력 회수가 1회줄어든다
IF CFLAG:MASTER:현재위치 == 99
	SETCOLOR C_RED
	PRINTFORMW ### COM_END LOCATION ERROR!%마스터는% 존재하지 않아야 할 장소에 있다!###
	PRINTW 강제적으로 하루를 종료합니다. 로그를 취해 스레에 버그 보고를 부탁합니다.
	RESETCOLOR
	CFLAG:MASTER:현재위치 = CFLAG:MASTER:311
	CALL SET_SLEEP(1, MASTER, 0)
	TFLAG:106 = 0
	BEGIN AFTERTRAIN
ENDIF
	;CFLAG:MASTER:취침시간 = DAY * 1440 + TIME
SIF SELECTCOM == 400
	TWAIT 100,0

@ENKAI_SINKOU
;연회의 참가인수를 세트
FLAG:35 = 0
FOR LOCAL, 0, CHARANUM
	IF CFLAG:LOCAL:20 == 1 || CFLAG:LOCAL:20 == 2
		SIF CFLAG:LOCAL:현재위치 == TFLAG:연회장
			FLAG:35 ++
	ENDIF
NEXT
;날씨가 나쁘면 중지가 되는 연회도 있다
IF FLAG:연회개최플래그 > 0 && TIME:5 >= 4 && FLAG:36 > 0 && DAY >= FLAG:34 && TIME >= FLAG:33
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		PRINTL
		PRINTW 유감스럽지만, 날씨가 나쁘기 때문에 연회는 중지가 되었습니다
	ELSE
		PRINTL
		PRINTW 유감스럽지만, 악천후 때문에 연회는 중지가 된 것 같습니다
	ENDIF
	CALL ENKAI_STOP
;연회의 도중에 내리기 시작했을 경우
ELSEIF FLAG:연회개최플래그 >= 98 && TIME:5 >= 4 && FLAG:36 > 0
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		PRINTL
		PRINTW 유감스럽지만, 날씨가 나쁘기 때문에 연회는 중지가 되었습니다
	ELSE
		PRINTL
		PRINTW 유감스럽지만, 악천후 때문에 연회는 중지가 된 것 같습니다
	ENDIF
	CALL ENKAI_STOP
;연회 개시
ELSEIF FLAG:연회개최플래그 > 0 && FLAG:연회개최플래그 != 98 && FLAG:연회개최플래그 != 99 && DAY >= FLAG:34 && TIME >= FLAG:33
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		IF FLAG:35 < 2
			PRINTL
			PRINTW 조용한 분위기안, 연회가 시작되었습니다…
			PRINTL
		ELSEIF FLAG:35 >= 8
			PRINTL
			PRINTW 떠들썩한 분위기안, 연회가 시작되었습니다…
			PRINTL
		ELSE
			PRINTL
			PRINTW 부드러운 분위기안, 연회가 시작되었습니다…
			PRINTL
		ENDIF
		FLAG:연회개최플래그 = 99
	ELSE
		FLAG:연회개최플래그 = 98
	ENDIF
ELSEIF FLAG:연회개최플래그 > 0 && FLAG:연회개최플래그 != 99 && DAY >= FLAG:34 && TIME >= FLAG:33
	IF CFLAG:MASTER:현재위치 == TFLAG:연회장
		PRINTL
		PRINTW 이미 연회는 시작되어 있는 것 같습니다…
		PRINTL
		FLAG:연회개최플래그 = 99
	ENDIF
	;주최자측은 아직 연회준비 플래그 인 채의 모아 두어 플래그를 억지로 진행한다
	FOR LOCAL, 0, CHARANUM
		SIF CFLAG:LOCAL:20 == 2
			CALL ENKAI_WORK(LOCAL)
	NEXT
ENDIF


code:
@EVENTTURNEND
#DIM 양육비

;연회 당일에 하루가 끝나면 연회 플래그 OFF
;IF DAY >= FLAG:34 && FLAG:연회개최플래그
IF TIME + 1440 * DAY >= FLAG:개시시간 + 1440 * FLAG:개시일 && FLAG:연회개최플래그
	;회장이 된 장소는 몹시 황폐해지는, 거점외에서도 대상 지점의 더러운 증가시키기 (위해)때문에 연회장은 아니다
	SIF YOGORE:(FLAG:32) < 300
		YOGORE:(FLAG:32) += 1000
	FLAG:연회개최플래그 = 0
	FLAG:31 = 0
	FLAG:32 = 0
	TFLAG:연회장 = 0
	FLAG:33 = 0
	FLAG:34 = 0
	FLAG:35 = 0
	FLAG:36 = 0
	;연회회장표시 =
	FLAG:연회차입 = 0
	FOR LOCAL,0,CHARANUM
		CFLAG:LOCAL:350 = 0
		MAXBASE:LOCAL:일량 = 0
		BASE:LOCAL:일량 = 0
		CFLAG:LOCAL:연회참가 = 0
		CFLAG:LOCAL:행동 = 0
	NEXT
ENDIF
;사당 플래그 OFF
FLAG:17 = 0
DRAWLINE
PRINTFORML
;SETCOLOR C_PINK
;FONTBOLD
FOR LOCAL,0,CHARANUM
	;この3つはあなたも取得可能
	IF EXP:LOCAL:키스경험 > 300 && ABL:LOCAL:M감각 > 5 && !TALENT:LOCAL:키스마
		PRINTFORMW %CALLNAME:LOCAL%는 [키스마]를 얻었다
		TALENT:LOCAL:키스마 = 1
	ENDIF
	IF EXP:LOCAL:조교자위경험 > 200 && EXP:LOCAL:자위경험 > 500 && !TALENT:LOCAL:자위중독
		PRINTFORMW %CALLNAME:LOCAL%는 [자위중독]을 얻었다
		TALENT:LOCAL:자위중독 = 1
	ENDIF
	IF EXP:LOCAL:자궁구개발경험 > 50 && !TALENT:LOCAL:포르치오성감
		PRINTFORML 반복해 자궁구를 자극받은 %CALLNAME:LOCAL%는 고통보다 쾌락을 느끼게 된 것 같다…
		PRINTFORMW %CALLNAME:LOCAL%는 [포르치오성감]을 얻었다
		TALENT:LOCAL:포르치오성감 = 1
	ENDIF
	;以下はあなたはスキップ
	SIF LOCAL == MASTER
		CONTINUE
	;연모 취득
	IF 연모취득조건(LOCAL)
		;면영기는 특별 취급
		IF LOCAL == 57
			PRINTFORMW %CALLNAME:LOCAL%는 %CALLNAME:MASTER%에의 마음을 자각함과 동시에, 입술끝에 미소를 머금었다
			PRINTFORMW 자연스러운 웃는 얼굴을 띠는 %CALLNAME:LOCAL%는 표정을 얻은 같다
			PRINTFORMW %CALLNAME:LOCAL%의 \@ TALENT:LOCAL:사모 ? [사모]는 # [애욕]은 \@ [연모]가 되었다
			FLAG:10 ++
		;애욕의 유무로 분기
		ELSEIF TALENT:LOCAL:애욕
			PRINTFORMW %CALLNAME:LOCAL%는 몸도 마음도 %CALLNAME:MASTER%에 채워지는 것에 즐거움을 느끼고 있다…
			PRINTL 문란한 관계 끝에 여문 연정, 그것도 또 하나의 사랑의 형태…
			PRINTFORMW %CALLNAME:LOCAL%의 [애욕]은 [연모]가 되었다
		ELSE
			PRINTFORML %CALLNAME:LOCAL%는 %마스터와% 보내는 시간에 행복을 느끼고 있는 것 같다…
			PRINTFORMW %CALLNAME:LOCAL%의 [사모]는 [연모]가 되었다
			FLAG:10 ++
		ENDIF
		SIF !MARK:LOCAL:시간각인 && !TALENT:LOCAL:애욕 && !CFLAG:LOCAL:수면간 && !CFLAG:LOCAL:만취간
			CFLAG:LOCAL:깨끗한교제 = 1
		TALENT:LOCAL:연모 = 1
		TALENT:LOCAL:사모 = 0
		TALENT:LOCAL:애욕 = 0
		TALENT:LOCAL:섹프 = 0
		CALL KOJO_MESSAGE_SEND("EVENT",21,LOCAL,1,0)

		;楽譜入手処理 ;KR판 악보기능 이식필요
		PRINTL
		; RESULT = 楽譜入手_陥落(LOCAL)
		; SIF RESULT != -1
		; 	PRINTL
	;사모 취득
	;섹프/정부의 취득으로 분기
	ELSEIF 사모취득조건(LOCAL)
		IF TALENT:LOCAL:섹프
			PRINTFORML %CALLNAME:LOCAL%는 %CALLNAME:MASTER%의 몸 외에 대해서도 흥미가 있는 것 같다
			PRINTFORMW %CALLNAME:LOCAL%는 [사모]를 얻었다
		ELSE
			PRINTFORML %CALLNAME:LOCAL%는 %CALLNAME:MASTER%에 대한 것이 신경이 쓰여요 다…
			PRINTFORMW %CALLNAME:LOCAL%는 [사모]를 얻었다
		ENDIF
		TALENT:LOCAL:사모 = 1
		CALL KOJO_MESSAGE_SEND("EVENT",21,LOCAL,2,0)
	;애욕 취득
	;봉사 쾌락의 분기가 아직 없기 때문에, 취득 조건에 병합
	ELSEIF 애욕취득조건(LOCAL)
		;사모 가져, 진심으로 몸을 선택한적인
		IF TALENT:LOCAL:사모
			PRINTFORML %CALLNAME:LOCAL%는 %마스터와% 마음 이상으로 몸의 연결에 만족한 것 같다…
			PRINTFORML 눈에 보이지 않는 정보다 눈에 보이는 증거를 바라 앞으로도 몸을 겹치는 것일까
			PRINTFORMW %CALLNAME:LOCAL%는 [애욕]을 얻었다
		ELSE
			PRINTFORML %CALLNAME:LOCAL%는 %마스터와%의 정욕에 빠지고 있는 것 같다…
			PRINTFORMW %CALLNAME:LOCAL%는 [애욕]을 얻었다
		ENDIF
			FLAG:10 ++
			TALENT:LOCAL:애욕 = 1
			TALENT:LOCAL:사모 = 0
			CALL KOJO_MESSAGE_SEND("EVENT",21,LOCAL,3,0)

			;楽譜入手処理 ;KR판 악보기능 이식필요
			PRINTL
			; RESULT = 楽譜入手_陥落(LOCAL)
			; SIF RESULT != -1
			; 	PRINTL
	;섹프/정부 취득
	;섹프와 정부의 차이는 사모의 유무
	ELSEIF 섹프취득조건(LOCAL)
		;사모 취득제＋섹프미취득
		IF TALENT:LOCAL:사모 && !TALENT:LOCAL:섹프
			PRINTFORML %CALLNAME:LOCAL%는 %마스터와% 함께라면 쾌락에 빠져도 괜찮다고 생각하기 시작한 것 같다…
			PRINTFORMW %CALLNAME:LOCAL%는 %CALLNAME:MASTER%의 [정부]가 되었다
			TALENT:LOCAL:섹프 = 2
			CALL KOJO_MESSAGE_SEND("EVENT",21,LOCAL,4,0)
		;사모 취득제＋섹프 취득제
		ELSEIF TALENT:LOCAL:사모 && TALENT:LOCAL:섹프 == 1
			PRINTFORML %CALLNAME:LOCAL%는 %CALLNAME:MASTER%에게 단순한 육체 관계 이상의 것을 바라고 있는 것 같다…
			PRINTFORMW %CALLNAME:LOCAL%는 [정부]가 되었다
			TALENT:LOCAL:섹프 = 2
			CALL KOJO_MESSAGE_SEND("EVENT",21,LOCAL,4,0)
		;사모미취득＋섹프미취득이라면 우선 섹프 취득
		ELSEIF !TALENT:LOCAL:섹프
			PRINTFORML %CALLNAME:LOCAL%는 %마스터가% 주는 쾌락에 빠져 있다…
			PRINTFORMW %CALLNAME:LOCAL%는 %CALLNAME:MASTER%의 [섹프]가 되어버렸다
			TALENT:LOCAL:섹프 = 1
			CALL KOJO_MESSAGE_SEND("EVENT",21,LOCAL,5,0)
		ENDIF
	ENDIF
	IF EXP:LOCAL:연주경험 >= 1000 && ABL:LOCAL:기교 > 3 && !TALENT:LOCAL:음악지식
		PRINTFORMW %CALLNAME:LOCAL%는 악기 연주의 조예가 깊어졌다
		PRINTFORMW %CALLNAME:LOCAL%는 [악기지식]을 얻었다
		;PRINTFORMW %CALLNAME:LOCAL%는 음악의 조예가 깊어졌다
		;PRINTFORMW %CALLNAME:LOCAL%는 [음악지식]을 얻었다
		TALENT:LOCAL:음악지식 = 1
	ELSEIF EXP:LOCAL:연주경험 >=5000 && !TALENT:LOCAL:음감
		PRINTFORMW %CALLNAME:LOCAL%의 음감은 꽤 단련할 수 있어 온 것 같다
		PRINTFORMW %CALLNAME:LOCAL%는 [상대 음감]을 얻었다
		TALENT:LOCAL:음감 = 1
	ENDIF
	IF CFLAG:LOCAL:언짢음
		CFLAG:LOCAL:언짢음 --
		IF !CFLAG:LOCAL:언짢음
			SIF MARK:LOCAL:반발각인
				CFLAG:LOCAL:언짢음 ++
		ENDIF
	ENDIF
	;EX캐릭터의 연모시는 멤버 추가 이벤트 플래그를 세운다
	IF !CFLAG:LOCAL:신사거주
		SIF TALENT:LOCAL:연모 && EXP:LOCAL:애정경험 >= 200
			FLAG:1 = 3
	ENDIF
NEXT

IF EXP:MASTER:이상청소경험 > (500 + 100 * FLAG:850) && ABL:MASTER:청소기능 > 3 && !TALENT:MASTER:청소중독
	PRINTFORMW %마스터는% 무서운 것에 깨닫아 버렸다
	PRINTFORMW 항상, 시간을 멈추고 있는 동안도, 소녀들과 살을 맞대고 있을 때조차,
	PRINTFORMW 방이 어질러져 있지 않은지 신경 쓰여 못 견디는 것이다
	PRINTFORMW 이미 %CALLNAME:MASTER%의 마음이 편안해지는 것은 이렇게 해 청소를 할 때뿐
	PRINTFORMW %CALLNAME:MASTER%의 청소에의 조건은, 돌이킬 수 없는 경지에 이르고 있었다
	PRINTFORMW %마스터는% [청소중독]을 얻었다
	TALENT:MASTER:청소중독 = 1
ENDIF
;MASTER를 제외한 전원의 pre-set 처리
FOR LOCAL,1,CHARANUM
	;기력의 회복
	BASE:LOCAL:기력 = MAXBASE:LOCAL:기력

	LOCAL:1 = MAXBASE:LOCAL:체력 / 3

	;회복 빠름, 회복 느림
	LOCAL:1 = LOCAL:1 * 3 / (3 - TALENT:LOCAL:회복속도)

	;동침하고 있으면 회복속도 향상
	SIF CFLAG:LOCAL:동침중
		LOCAL:1 *= 2
	SIF CFLAG:LOCAL:철야
		LOCAL:1 /= 2
	CFLAG:LOCAL:철야 = 0
	BASE:LOCAL:체력 = MIN(MAXBASE:LOCAL:체력, BASE:LOCAL:체력 + LOCAL:1)

	;나무 삼개 드 발행자가 있으면 발행 플래그 OFF
	CFLAG:LOCAL:328 = 0
NEXT
;전원의 pre-set 처리
FOR LOCAL, 1, CHARANUM
	BASE:LOCAL:정력 = MIN(MAXBASE:LOCAL:정력, BASE:LOCAL:정력 + MAXBASE:LOCAL:정력 * BASE:LOCAL:체력 / MAXBASE:LOCAL:체력)

	;은밀중의 캐릭터가 있으면 플래그 OFF
	CALL SET_SNEAK(0,LOCAL,1)
	;CFLAG:LOCAL:은밀중 = 0

	;장난중의 캐릭터가 있으면 플래그 OFF
	CALL SET_PRANK(0,LOCAL,1)
	;CFLAG:LOCAL:장난 = 0

NEXT

;연회의 개최 판정
SIF FLAG:3 && FLAG:연회개최플래그 == 0
	CALL ENKAI_SETTING

IF CFLAG:MASTER:육아인원수
	양육비 = CFLAG:MASTER:육아인원수 * 300 / (1 + FLAG:생활능력무) ;원판은 인당 1000원
	PRINTFORML %마스터는% 아이의 양육비로서 \\{양육비} 지불했다.
	MONEY -= 양육비
ENDIF

CALL RANK_POINT
PRINTW

FLAG:0 = 0

;모브코의 세이브
IF TALENT:RANDOM_CHARANUM:스침 == 1 && GETBIT(TFLAG:일일일회, 9)
	PRINTFORML %마스터는% 오늘 만난 %CALLNAME:RANDOM_CHARANUM%를……
	CALL ASK_YN("잊는다","기억해 둔다")
	IF RESULT
		VARSET TCVAR:RANDOM_CHARANUM:0, 0
		VARSET EX:RANDOM_CHARANUM:0, 0
		VARSET TEQUIP:RANDOM_CHARANUM:0, 0
		VARSET PALAM:RANDOM_CHARANUM:0, 0
		CALL InitializeBeforeTrain(RANDOM_CHARANUM)
		CFLAG:RANDOM_CHARANUM:900 ++
		CALL SAVE_MOBGIRL
		SIF RESULT == -1
			GOTO FORGET_MOB
		PRINTFORMW %마스터는% %CALLNAME:RANDOM_CHARANUM%에 또 만날 수 있으면 좋겠다라고 생각했다
	ELSE
		$FORGET_MOB
		PRINTFORMW %마스터는% %CALLNAME:RANDOM_CHARANUM%를 완전히 잊었다
	ENDIF
ENDIF
IF TALENT:RANDOM_CHARANUM:스침 == 2 && GETBIT(TFLAG:일일일회, 9)
	CALL MOBGIRL_PROFILE2()
	VARSET TCVAR:RANDOM_CHARANUM:0, 0
	VARSET EX:RANDOM_CHARANUM:0, 0
	VARSET TEQUIP:RANDOM_CHARANUM:0, 0
	VARSET PALAM:RANDOM_CHARANUM:0, 0
	CALL InitializeBeforeTrain(RANDOM_CHARANUM)
	SAVECHARA TOSTR(CFLAG:RANDOM_CHARANUM:모브코세이브슬롯), CSTR:RANDOM_CHARANUM:모브코프로필1, RANDOM_CHARANUM
	[IF DEBUG]
		PRINTFORMW %CALLNAME:RANDOM_CHARANUM%の情報を更新しました。
	[ENDIF]
ENDIF

BEGIN SHOP

@RANK_POINT
VARSET LOCAL
;청결도
LOCAL:2 = EX:MASTER:오늘의청소회수 / 10
PRINTFORML 오늘 청소한 쓰레기의 양 {EX:MASTER:오늘의청소회수}
PRINTFORML 청소로 얻은 보너스 \\{LOCAL:2}
IF DAY:2 == 4 && DAY:3 == 31
	PRINT (x2)  【대청소 보너스】
	LOCAL:2 *= 2
ENDIF
;거점의 주인의 호감도와 신뢰도가 상승
LOCAL:3 = MIN(10, EX:MASTER:오늘의청소회수 / 2000)
IF LOCAL:3
	PRINTFORML 청소를 열심히 한 덕분에 %CALLNAME:거점_주인%로부터의 평가가 올랐다
	CFLAG:거점_주인:호감도 += LOCAL:3 * 2
	CFLAG:거점_주인:신뢰도 += LOCAL:3
ENDIF
PRINTL
;심부름 보수
LOCAL:4 = TCVAR:MASTER:도운양

SIF LOCAL:4 > 0
	PRINTFORML 업무 도우미로 얻은 보너스 \\{LOCAL:4}

MONEY += LOCAL:2 + LOCAL:4

;호감도 상승
IF FLAG:12 && MONEY:2 < CM_LIMIT
	LOCAL:5 = MAX(1,FLAG:12 / 100)
	LOCAL:5 = MIN(LOCAL:5, CM_LIMIT - MONEY:2)
	PRINTFORML 호감도 상승 보너스 ＋{LOCAL:5}카리스마
	MONEY:2 += LOCAL:5
	PRINTFORML 현재의 카리스마：{MONEY:2}　카리스마 보유 상한：{CM_LIMIT}
ENDIF
FLAG:12 = 0

PRINTW

@연모취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:연모
	RETURNF 0
SIF CFLAG:ARG:호감도 < REQUIRED_FAVOR_연모
	RETURNF 0
SIF CFLAG:ARG:신뢰도 < REQUIRED_TRUST_연모
	RETURNF 0
SIF ABL:ARG:친밀 < REQUIRED_AFFECTION_연모
	RETURNF 0
SIF MARK:ARG:반발각인
	RETURNF 0
SIF TALENT:ARG:애욕 && !FLAG:함락노선변화
	RETURNF 0
RETURNF 1

@사모취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:애욕
	RETURNF 0
SIF TALENT:ARG:연모
	RETURNF 0
SIF TALENT:ARG:사모
	RETURNF 0
SIF CFLAG:ARG:호감도 < REQUIRED_FAVOR_사모
	RETURNF 0
SIF CFLAG:ARG:신뢰도 < REQUIRED_TRUST_사모
	RETURNF 0
SIF ABL:ARG:친밀 < REQUIRED_AFFECTION_사모
	RETURNF 0
SIF TALENT:ARG:섹프 && !FLAG:함락노선변화
	RETURNF 0
RETURNF 1

@애욕취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:연모
	RETURNF 0
SIF TALENT:ARG:애욕
	RETURNF 0
SIF !TALENT:ARG:섹프
	RETURNF 0
SIF CFLAG:ARG:호감도 < REQUIRED_FAVOR_애욕
	RETURNF 0
SIF ABL:ARG:욕망 < REQUIRED_DESIRE_애욕
	RETURNF 0
SIF (EXP:ARG:절정경험 - EXP:ARG:무자각절정경험) < REQUIRED_ECSTACY_애욕
	RETURNF 0
SIF (ABL:ARG:욕망 - ABL:ARG:친밀) < 1
	RETURNF 0
RETURNF 1

@섹프취득조건(ARG)
#FUNCTION
SIF TALENT:ARG:연인
	RETURNF 0
SIF TALENT:ARG:연모
	RETURNF 0
SIF TALENT:ARG:섹프 == 2
	RETURNF 0
SIF CFLAG:ARG:호감도 <= REQUIRED_FAVOR_섹프
	RETURNF 0
SIF ABL:ARG:욕망 < REQUIRED_DESIRE_섹프
	RETURNF 0
SIF (EXP:ARG:절정경험 - EXP:ARG:무자각절정경험) < REQUIRED_ECSTACY_섹프
	RETURNF 0
SIF (ABL:ARG:욕망 - ABL:ARG:친밀) < 0
	RETURNF 0
RETURNF 1


code:
;신사라면, 역시 운세 뽑기죠
;그런 이유로 운세 뽑기 시간입니다
;현상은 뽑기만 하지 별다른 의미가 없습니다만(플래그는 장식이에요)
;LOCAL 0:전체운, 1:금전운, 2:건강운, 3:연애운, 4:여행운, 99:덤으로 한마디
;의 각 항목의 선택 번호.
;무엇을 뽑았는지 참조해서 나중에 처리할 수 있는 구조로 해두었기에, 추가하고 싶을 경우는 각 항목의 마지막에 덧붙여 씁시다
@FORTUNE
;우선은 제비의 결과를 보여드리죠
;전체운, 소재는 조만간 늘리겠습니다
PRINT 　전체운：
PRINTDATAW LOCAL:0
	DATAFORM 대길
	DATAFORM 중길
	DATAFORM 소길
	DATAFORM 길
	DATAFORM 반길
	DATAFORM 말길
	DATAFORM 말소길
	DATAFORM 흉
	DATAFORM 소흉
	DATAFORM 반흉
	DATAFORM 말흉
	DATAFORM 대흉
	DATAFORM
	DATAFORM 토길 (兎吉)
	DATAFORM 견길 (犬吉)
	DATAFORM 묘길 (猫吉)
	DATAFORM 조길 (釣吉)
	DATAFORM 당첨
	DATAFORM 꽝
	DATAFORM 오늘의 말린 오징어
	DATAFORM 광 (狂)
	DATAFORM 향 (鄕)
	DATAFORM 모후길
ENDDATA

PRINT 　금전운：
PRINTDATAW LOCAL:1
	DATAFORM 없는 지갑을 털지 말 것
	DATAFORM 낭비할 것 같다. 주의해야 할 것
	DATAFORM 임시수입을 들어오 예감 있음
	DATAFORM 견실이 제일
	DATAFORM 투자한다면 먼저 자기 자신에게 할 것
	DATAFORM 지갑끈을 조여라
	DATAFORM 돌고 도는 것, 사용하면 돌아온다
	DATAFORM 길을 잘못 들지 않으면 낭비는 길
	DATAFORM 낭비는 흉, 지금은 참을 때
	DATAFORM 잃어버린 것이 발견된다
	DATAFORM 좋은 이야기일수록 의심할 것
	DATAFORM 하면 이루어진다
	DATAFORM 해도 이루어지지 않는다
	DATAFORM 한 대로 요구해라
	DATAFORM 보답을 요구하지 말지어다
	DATAFORM 포기해라
	DATAFORM 마음 가는 대로 하라
	DATAFORM 왼쪽 매매 또한 재미
	DATAFORM 지식에 쏟아부을 것
	DATAFORM 습득물에 불행 있다
	DATAFORM 습득물에 행운 있다
	DATAFORM 흘린 물건에 불행 있다
	DATAFORM 흘린 물건에 행운 있다
	DATAFORM 1회 휴식
	DATAFORM 지불에 주의
	DATAFORM 수상쩍은 이야기에야말로 활로 있다
	DATAFORM 수상쩍은 이야기는 흉하다
	DATAFORM 뭐…… 힘내라
	DATAFORM 사치에 교만하지 말지어다
	DATAFORM 고간의 보물을 조심해라
ENDDATA

PRINT 　건강운：
PRINTDATAW LOCAL:2
	DATAFORM 오늘 밤은 일찍 자
	DATAFORM 밤샘도 때로는 보탬이 된다
	DATAFORM 일찍 일어나는 새가 벌레를 잡는다
	DATAFORM 일찍 일어나면 나쁜 소식을 듣는다
	DATAFORM 새끼발가락에 주의
	DATAFORM 책상 모서리를 경계하라
	DATAFORM 머리 위 주의
	DATAFORM 발 밑 주의
	DATAFORM 전방 주의
	DATAFORM 후방 주의
	DATAFORM 좌우 주의
	DATAFORM 동방 주의
	DATAFORM 서방 주의
	DATAFORM 북방 주의
	DATAFORM 남방 주의
	DATAFORM 중앙 주의
	DATAFORM 난입 주의
	DATAFORM 식사에 어려움 있다
	DATAFORM 식사는 모든 건강의 근원
	DATAFORM 술은 백약지장
	DATAFORM 술은 백독지장
	DATAFORM 마시고 삼키고 신나게 마실 것
	DATAFORM 마셔도 술에 먹히지 말지어다
	DATAFORM 될 대로 된다
	DATAFORM 될 대로 안 된다
	DATAFORM 될 대로밖에 안 된다
	DATAFORM 될 대로도 안 된다
	DATAFORM 문제 없음
	DATAFORM 문제뿐. 경계를 게을리하지 말도록
	DATAFORM 칼부림 사태를 주의하라
	DATAFORM 사소한 계기가 신세를 망친다
	DATAFORM 사소한 계기가 신세를 부지한다
	DATAFORM 희망은 버리지 말지어다
	DATAFORM 무모한 짓은 새로운 문을 연다
	DATAFORM 무모한 짓은 하지 않고 안정을
	DATAFORM 주워먹는 짓은 삼가라
	DATAFORM 먹는 것에 불안 없다
	DATAFORM 차린 밥상은 소녀의 고집
	DATAFORM 치린 밥상을 먹지 않는 건 고자뿐
	DATAFORM 매일의 훈련을 빠뜨리지 말 것
	DATAFORM 과도한 훈련은 몸을 망친다
	DATAFORM 쿵후가 부족합니다
	DATAFORM 쿵후가 충분합니다
ENDDATA

PRINT 　연애운：
PRINTDATAW LOCAL:3
	DATAFORM 자신이 믿은 길을 뚫고 나아가라
	DATAFORM 믿어 돌진한 미래가 완전하게 예상외로 사랑의 소망을 끊기고 단절 상태가 보내져 오는 것은
	DATAFORM 지금은 그때가 아니다
	DATAFORM 지금이지
	DATAFORM 언제냐니 지금이지
	DATAFORM 동성에게 인기 있을 조짐
	DATAFORM 이성에게 끌릴 조짐
	DATAFORM 다만 진행해 단지 진행하고
	DATAFORM 급돌아라
	DATAFORM 부딪쳐서 깨져라
	DATAFORM 부딪치면 깨진다
	DATAFORM 지금은 뽑아
	DATAFORM 지금은 눌러
	DATAFORM 이루어지지 않고, 신식연을 찾을 것
	DATAFORM 이별은 만남의 시작이라
	DATAFORM 만남은 이별의 시작이라
	DATAFORM 깨닫지 못한 새에 사랑은 여문다
	DATAFORM 포기해
	DATAFORM 원점으로 되돌아 간다
	DATAFORM 뭐…… 화이팅, 이야
	DATAFORM 봄이 오다
	DATAFORM 여름이 오다
	DATAFORM 질려오다
	DATAFORM 겨울이 오다
	DATAFORM 새 봄이 왔다
	DATAFORM 호의의 뒤를 의심하지 말 것
	DATAFORM 호의의 뒤를 의심할 것
	DATAFORM 본심을 털어놓아, 말하지만 좋아
	DATAFORM 배의 나를 찾지 말 것
	DATAFORM 본심을 털어놓으면 배 할애해진다
	DATAFORM 비 내리지 않고와도 땅 굳어진다
	DATAFORM 비라고 땅 굳어지지 않고
	DATAFORM 그 사랑은 환상에 지나지 않는, 현실을 다시 봐라
	DATAFORM 그 사랑은 이차원에 지나지 않는, 삼차원으로 돌아간다
	DATAFORM 지금은 알 때
	DATAFORM 지금은 히나 때
	DATAFORM 자립 때야말로 가장 경계할 때
	DATAFORM 자립한 새는 둥지에 있지 않고
	DATAFORM 자립한 새는 둥지로 돌아간다
	DATAFORM 마음은 전해진다
	DATAFORM 마음은 비뚤어진다
	DATAFORM 마음은 전해지지 않고
	DATAFORM 마음의 일그러짐은 인연도 비뚤어지게 한다
	DATAFORM 사랑은 종족조차 넘어선다
	DATAFORM 종족의 벽 두텁도다
	DATAFORM 욕망은 이성을 넘는다
	DATAFORM 이성은 욕망을 억누른다
	DATAFORM 사랑은 이성을 녹인다
	DATAFORM 이성은 사랑을 억누른다
	DATAFORM 욕망은 사랑을 변질시킨다
	DATAFORM 사랑은 욕망을 맑게 개이게 한다
	DATAFORM 사랑과 욕망이 손을 잡았다. 바라는 대로 마음을 성취하라
	DATAFORM 상대를 돌보는 형태가 좋다
	DATAFORM 상대를 억제하는 형태가 좋다
	DATAFORM 당신을 억제하는 형태가 좋다
	DATAFORM 자신의 생각하는 대로 가는 것이 길하다
	DATAFORM 상대에게 맡기는 것이 좋아
	DATAFORM 마음은 몸에 딸려간다
	DATAFORM 몸은 마음에 딸려간다
	DATAFORM 육욕이 가는 대로
	DATAFORM 육욕에 내맡기는 것은 재앙의 화근
ENDDATA

PRINT 　여행운：
PRINTDATAW LOCAL:4
	DATAFORM 견문을 넓히는 여행을 떠나라
	DATAFORM 집에 있는 것이 길하다
	DATAFORM 인간마을에 흉조
	DATAFORM 인간마을에 길조
	DATAFORM 오랜 유적이 좋다
	DATAFORM 오랜 유적에 다가가지 말라
	DATAFORM 절은 좋은 기로 가득하다
	DATAFORM 절에 침전물 서로 봐
	DATAFORM 안개 강하게 흐린다
	DATAFORM 안개 개인다
	DATAFORM 붉은 저택에 좋은 꿈 있음
	DATAFORM 붉은 저택에 악몽 있음
	DATAFORM 대나무 끝에 빛 있음
	DATAFORM 대나무 속에 미혹 있음
	DATAFORM 숲에 가고는 양식
	DATAFORM 숲에 행 재앙 있어
	DATAFORM 저승에 좋은 서로 불러
	DATAFORM 저승에 초대되고 바구니점괘 있어
	DATAFORM 산으로 향하는 게 좋다
	DATAFORM 산으로 향하지 않는 게 좋다
	DATAFORM 산 꼭대기는 쾌청하다
	DATAFORM 산 꼭대기는 황천하다
	DATAFORM 땅밑은 고요하다
	DATAFORM 땅밑이 소란스럽다
	DATAFORM 예정을 세우면 걱정 없음
	DATAFORM 하늘을 가는 것이 길하다
	DATAFORM 생각하지만 진에 가지만 좋다
	DATAFORM 조난하지만 살아남는다
	DATAFORM 예정을 생각하지 않고 가는 것도 좋다
ENDDATA

PRINT 　한마디：
PRINTDATAW LOCAL:99
	DATAFORM 당첨, 한 장 더
	DATAFORM 네, 한 번 더
	DATAFORM 1주간 뒤에 한 번 더 뽑으러 오세요. 진정한 운세 뽑기를 보여드리겠습니다
	DATAFORM 그대는 이 결과를 믿어도 되고, 내일도 한 번 더 뽑아도 된다
	DATAFORM 믿는 것도 믿지 않는 것도 당신하기 나름
	DATAFORM 믿는 자는 구원받는다
	DATAFORM 결과에 불만이라면 내일 또 와주십시오
	DATAFORM ……뒤에 (문장은 여기서 끊겨 있다)
	DATAFORM 액을 털어내고 싶으면 지금 당장 새전을!
	DATAFORM 고민을 털어놓는 것도 때로는 필요합니다
	DATAFORM 마음은 그때가 올 때까지 숨겨두자
	DATAFORM 환상향　아아 환상향　환상향
	DATAFORM 이 운세를 읽었을 때, 과연 나는 이 세상에 아직 남아있을까요?
	DATAFORM 소재 내놔, 얼렁
	DATAFORM 덧붙여 당 신사의 새전함에는 아직 약간의 여유가 있습니다
	DATAFORM b의 f말할 수 있는Vhr, c2m0응fh4；l(이해를 거부하게 하는 뭔가가 적혀 있었다)
	DATAFORM 맞는 것도 팔괘, 안 맞는 것도 팔괘
	DATAFORM ↑↑↓↓←→←→봄샤ㅅ(피츙―)
	DATAFORM 진척 어때요?
	DATAFORM 진척 안 돼요
ENDDATA

;확인용. 각각 몇 번째의 것이 놓쳤는지 콘솔에 기록
;DEBUGPRINTFORML FORTUNE:{LOCAL:0, 2}, {LOCAL:1, 2}, {LOCAL:2, 2}, {LOCAL:3, 2}, {LOCAL:4, 2}, {LOCAL:99, 2}

;현상은 표시만
;운세 결과에 응해 어떠한 액션을 일으킨다면 이쪽을 묻어 주세요
;각 항목의 n-1행째가 각 변수에 격납되고 있습니다

SELECTCASE LOCAL:0
	CASE 0,19
		TFLAG:행운보정 = 5
	CASE 13 TO 15
		TFLAG:행운보정 = 2
		TFLAG:귀보정 = LOCAL:0 - 11
	CASE 1 TO 3, 16, 17
		TFLAG:행운보정 = 3
	CASE 4 TO 6
		TFLAG:행운보정 = 1
	CASE 7 TO 10
		TFLAG:행운보정 = -3
	CASE 11,18
		TFLAG:행운보정 = -5
	CASEELSE
ENDSELECT

SELECTCASE LOCAL:1
CASEELSE
ENDSELECT

SELECTCASE LOCAL:2
CASEELSE
ENDSELECT

SELECTCASE LOCAL:3
CASEELSE
ENDSELECT

SELECTCASE LOCAL:4
CASEELSE
ENDSELECT

SELECTCASE LOCAL:99
	;맞았으므로 이제(벌써) 한 장(다음의 제비를 뽑았을 때에 캐쉬 백)
	;다만, 정기 계약중 이라면 캐쉬 백 이야기(처리가 귀찮아서)
	CASE 0 TO 2
		SIF !FLAG:운세뽑기계약
			SETBIT FLAG:999, 10
CASEELSE
ENDSELECT

RETURN 0


code:
;	GIFT.ERB
;		Version 0.03-0. 5
;		원래 proto판의 바리안트에 더욱 proto판의 패치를 만든다든가 무엇 생각하는거야 이녀석, 대용품
;		변수 A, B를 고쳐 쓸 뿐만 아니라 CFLAG:990을 마음대로 사용하고 있기 (위해)때문에, 본체의 버전 업에
;		따라 수정 필수
;		그것을 밟은 위에 당신은 이 패치를 도입해도 되고 도입하지 않아도 괜찮다
;		일단, 얻은 팬티의 소유자와 종류는 기록하고 있기 (위해)때문에, 후에 무언가에 사용할 수 있는 것 같은 생각이 들지 않는 것도 아니다
;		현상, 관리가 귀찮아서 당신이 취침중 때에 취침중이 아니었다 노예 마셔 실행 가능, 이라고 한다
;		묶음새가 미실장
;
;		패치 작성이라든지 처음에 어째서 소스가 더러운데는 용서를(·ω·｀
;
;		사용 변수
;			A, B
;
;		사용 CFLAG
;			990	선물 된 팬티의 취득 판정에 이용한다. 노예 옆으로부터는 「선물 한 팬티의 종류」라고 말하는 취급
;				본체의 (분)편으로 이용하게 되면(자) 확실히 번호 변경
;
;		향후 실장 예정
;			어제 신고 있던 것을 선물, 그게 아니라, 지금 신고 있는 것을 벗어 그 자리에 남겼다는 판도 만들고 싶다
;
;		중요한 말
;			예정은 미정
;
;		-- 이력 --
;			2010/07/18	ProtoVer0. 03용 0.05 공개
;
@GET_GIFT
VARSET LOCAL
LOCAL = RAND:(CHARANUM-1) + 1
;호감도 1000이상, 트렁크스 외의 속옷을 입고 있다
SIF CFLAG:LOCAL:2 < 1000 || EQUIP:LOCAL:6 < 2
	RETURN 0
SIF RAND:10 > 0
	RETURN 0

;묘사
DRAWLINE
PRINTFORMW %마스터가% 눈을 뜨니 머리맡에 뭔가 눈에 익지 않은 것이 놓여 있었다
IF RAND:3 == 2
	PRINTFORM 난잡하게
ELSEIF RAND:2 == 1
	PRINTFORM 정중하게
ELSE
	PRINTFORM 대충
ENDIF
PRINTFORM 개어둔 그것을 펼쳐보니, 그것은 아무래도 %CALLNAME:LOCAL%가 소유한
PRINTFORMW %PANTSNAME(EQUIP:LOCAL:6, LOCAL)% 같다
PRINTFORMW 미묘하게 주름이 남은 그것은, 아무래도 %CALLNAME:LOCAL%가 하루 신은 뒤의 것임이 사
PRINTFORMW %마스터는% 그것을 책상의 서랍에 넣고는, 아무 일도 없었던 것처럼 몸치장을 개시했다

PRINTFORML
PRINTFORMW %CALLNAME:LOCAL%의 %PANTSNAME(EQUIP:LOCAL:6, LOCAL)%를 입수했습니다
PRINTFORML

[SKIPSTART]
LOCALS = %CALLNAME:LOCAL%{EQUIP:LOCAL:6}/
SIF SAVESTR:10 == ""
	SAVESTR:10 = /
SAVESTR:10 += LOCALS
[SKIPEND]
CFLAG:LOCAL:(EQUIP:LOCAL:6) += 1


code:
#DIM UFUFU_START_TIME
#DIM UFUFU_START_DAY